
Session 01: Overview of CSPro, Dictionary and Forms

At the end of this session participants will be able to:

¢ |dentify different CSPro modules and tools and their roles in the survey workflow
o Create a simple data entry application including dictionary and forms

e Run a data entry application on Windows

e Run a data entry application on mobile and retrieve the data entered

¢ View data using Data Manager

CSPro Overview
CSPro is a suite of software tools for census and survey data processing that includes modules for data collection, editing,
tabulation, and dissemination.

CSPro has a long history. It was first released in 2000 and has been used in over 100 countries worldwide. It has been used
for censuses all over the world as well as for many large and complex household surveys including the Demographic and
Health Survey (USAID), Multiple Indicator Cluster Survey (UNICEF) and Living Standards Measurement Study (World Bank).
The first Android version was released in 2014. CSPro Android has been used in production for household surveys and
population censuses in multiple countries.

CSPro is free software developed by the US Census Bureau and funded by USAID. The Census Bureau provides free email
customer support. You can send questions to cspro@ lists.census.gov. There is also a user forum where you can post
questions and see answers to questions posted by other users at https://www.csprousers.org/forum.

CSPro can be used for both the traditional PAPI (pencil and paper interview) workflow as well as the computer aided
personal interview (CAPI) workflow. In this workshop, we will focus on data collection in CSPro using a CAPI workflow.

Required Hardware and Software
To run the examples in this workshop you will need the following:

e Windows PC with CSPro version 8.0 or later

e Android tablet with CSEntry version 8.0 or later

Using CSPro Android
Before we learn how to build data entry applications in CSPro, let's try doing some data collection to get comfortable with
the system.

Group Exercise
Split into groups of 3-4 people and use the provided tablets to interview each other using the "Getting to Know You"
application. Interview each member of the group so that we have data for all workshop participants. When you are

e
done, tap the sync button () to upload your results to the server.

Creating a Data Entry Application

For most of the examples in this workshop, we will be creating a data entry application based on the Popstan 2020 Census
questionnaire included with the workshop materials. With CAPI applications, a paper questionnaire alone is not sufficient to
define a data entry application. We also a need a specification document that describes consistency checks, skip patterns,
text fills, error messages and other aspects of the interactive application that are not defined on a paper questionnaire.
Take a moment to review both the questionnaire and the accompanying specification document.

While CSPro data entry programs can be run on Android and Windows tablets and phones, to develop a survey or census
application in CSPro you will use the CSPro Designer which runs on Windows PCs. When you launch CSPro Designer you are
given the choice of "Data Entry Application" for key from paper (PAPI) and "CAPI Data Entry Application" for electronic data
collection using phones/tablets/laptops. The differences are:

o System controlled (tightly controlled path)

e CAPI question text

e Extended controls (radio buttons, checkboxes, date picker, ...)

Since we are creating a CAPI application we will choose "CAPI Data Entry Application". We will name the application
"Popstan2020" and we will use the same name for the dictionary. Since we will eventually add other applications such as the
listing questionnaire and the menu, we will create the following folder structure:

The Data Dictionary
The first step in creating the application is to define the data dictionary. The data dictionary lists all the data items and

possible responses that will be in the application and organizes them in records and levels. The dictionary has the following
hierarchy:

Popstan

2020

Household

W Popstan2020.dcf
- Popstan2020.ent
&= Popstan2020.ent.apc
Ij Popstan2020.ent.mgf
Ij Popstan2020.ent.gsf
T Popstan2020.fmf

E

Value Sets

Before defining the record and items we first create ID items. ID items uniquely define each case (each questionnaire).
Usually these are geographic codes.

What are the ID items for the example questionnaire?

Province, District, Enumeration Area, Area Type, and Household Number.

Why not include GPS, interview date, start and end time since they are in the same section of the questionnaire? Because
they are not part of the codes needed to uniquely identify the questionnaire.

Add the ID items to the dictionary

Province (numeric, length 1)

District (numeric, length 2)
Enumeration area (numeric, length 3)
Area type (numeric, length 1)

Household number (numeric, length 3)

Properties of dictionary items:

Label: text displayed to interviewers (not the full question text, we will see where to put that in a later lesson)
Name: used to refer to variable in CSPro logic (interviewers will never see this)

Start: column number of first digit (character) of variable

Len: number of characters/digits used to store variable

Data type: alpha for names and other text, numeric for numbers and coded responses

Item type: whether variable is item or subitem

Occ: number of occurrences for variables that are repeated in record

Dec: for numbers with fractional parts, number of digits after decimal point

Dec char: whether or not to include decimal point in decimal numbers

Zero fill: for numeric values; whether to add zeros or blanks to left of number when number of digits in value is less
than the length of the variable. Using this will avoid problems when dealing with subitems. We can enable zero fill by
default on the options menu.

Tip: Note that you can toggle showing names or labels in the dictionary tree on the left side of the screen using the
View menu. You can also select "Append names to labels in tree" to show both at the same time.

What are the records used in our survey?
Record Type Record Name Section(s) of the Questionnaire

A

B
E
F
G

INTERVIEW_REC A. Non ID items (A6-A10)

PERSON_REC B. Demographics, C. Education, D. Fertility,
DEATHS_REC E. Deaths of Household Members in the Past 5 years
HOUSING_REC F. Housing Characteristics

POSSESSIONS_REC  G. Household Possessions

Note that we could have separate records for education and fertility but instead we will combine them with the person
record. This will simplify analysis later on since we will not have to link the records together. Later on, we will see that even
with the records combined we can still put education and fertility into separate rosters on our forms.

Let's start by just creating the housing and person records.

Properties of records:

Label/name: same as for variables.
Type Value: to distinguish between different records in the data file.
Required: whether or not the record must be entered for the questionnaire to be complete.

Max: for multiply occurring records the maximum number of occurrences allowed. Generally 1 for singly occurring
records (like housing) and a larger number for repeating records (like 50 for household members). Note that CSPro
doesn't allocate space in the data file for occurrences that are not used so it is better to err on the side of caution and
allow extra occurrences.

What are the properties of the housing record?

e Type value: F (can use anything but nice to use something meaningful like section letter)
e Required: no (we will not collect section F for vacant/refusal)

e Max:1

for the person record?

e Type value: B
e Required: no (we can have empty households)

e Max: 30 (questionnaire has limit of 10 but no penalty for adding a few extra just in case)

Now add some fields to the person record:

e Person number (numeric length 2)*
e Name (alpha length 30)

e Relationship (numeric length 1)

e Sex (numeric length 1)

e Age (numeric length 3)
And to the housing record:

e Number of rooms (numeric length 2)

e Type of main dwelling (numeric length 1)

A note on variable naming
Different people have different styles of naming dictionary variables. Some use a descriptive name such as
"PLACE_OF_BIRTH" others prefer to use the question number such as "B07" and others prefer a combination such as
"B0O7_PLACE_OF_BIRTH". Whichever approach you choose just make sure that it will be easy for users of your
application and your data to understand. Will everyone working on the logic for your application know what B07 is?

For each of our variables we need to add the possible responses (value sets). The value set lists all valid responses along with
their corresponding labels for coded variables. Without a value set, the interviewer can enter any value (except blank) but
with a value set they are limited to the options defined in the value set. Without a values set, users can even enter negative
numbers. For this reason, it is good practice to use a value set for all numeric variables.

Define the value sets for some of our variables based on the response codes on the questionnaire:

e Area type (1- Urban, 2- Peri-urban, 3- rural)

e Province (copy/paste from Excel)

e Person number (range: 1-30, use "1-30" as the label for the range)
e Name (no value set)

e Sex (1- Male, 2-Female)

e Relationship (see questionnaire)

e Age (use a range: 0-120, plus don't know code 999). We can set the entry for 999 to the special value "missing". This
tells CSPro that this is a special code indicating that the response is unknown. We will see later how we can take
advantage of this when doing consistency checks

e Number of rooms (use generate value set to generate the codes 1-20)

¢ Type of main dwelling (use codes from questionnaire)

1The line number is not needed in CSPro itselfas there are ways to determine the row number usinglogic, however, when exporting the data to other

packages itis often useful to have it. We will see later how to fill this in automatically during data entry.

Dictionary Macros

There are some useful functions for working with dictionaries that you can access by right-clicking on the dictionary in the
tree on the left side of the screen and choosing "Dictionary Macros". In particular you can copy/paste all value sets or all
item names/labels from the dictionary to/from Excel. This can be used to create codebooks to share with people who do not
have access to CSPro. It can also be used to do bulk modifications on dictionary items such as renumbering values in value
sets or adding prefixes to item names.

Forms
Before we can enter data, we need to create data entry forms. To start, click on the yellow stack of forms on the toolbar. To
follow the look of the paper questionnaire we will create one form for each page of the paper questionnaire.

Create a form for section A: Identification. Drag and drop the ID items onto the form. Note that we can drag and drop
individual items or entire records. By right clicking on the form in the forms tree on left side of the screen we can change the
label and name of the form. Let's make the label "A: Identification" and make the name "IDENTIFICATION_FORM".

Create a form for section B: Demographics. Drag drop the items from the person record. Let's give the form label "B:
Demographics" and name "DEMOGRAPHICS_FORM". Note that when we drop the record we have the option to put the
items in a roster or a repeating form. If we drop the items on the household identification form, we can only roster since the
household identification isn't repeated. For our example let's use a roster.

When we create the rosters, CSPro automatically gives them a name that ends in "000", for example "PERSON0OQ". You can
see this in the forms tree on the left side of the screen. We can change this by right clicking on the roster in the forms tree
and choosing properties. Let's name our roster "DEMOGRAPHICS_ROSTER".

Create a form for section: F: Housing Characteristics. Drag and drop the items from the housing record.

For paper and pencil surveys, we would spend a lot of time on the layout, adding additional labels and frames to make the
form look exactly like the questionnaire. However, when rendered on Android, the form is rendered one question at a time
so making the form look like the paper form is not as important.

Running the Application on Windows
Click on the green light icon on the toolbar to run the data entry application on your PC. Enter the name of the data file. For
data files CSPro uses the file extension ".csdb". Let's name our data file "Household.csdb".

Data File Types

By default CSPro stores data in a CSPro database file with the extension .csdb. This file not only contains the data itself but it
also contains the notes, the index, the partial save status and metadata used for data synchronization. CSPro database files
can be viewed by double clicking on them or using the Data Manager tool from the tools menu.

The csdb extension is new in CSPro 7.0. In earlier versions of CSPro, the notes, index and partial save were stored in extra
files that accompanied the data file itself. Unlike the text file used by earlier versions of CSPro, this is a binary file that cannot
be viewed using Text Viewer. While it is still possible to use text files in CSPro 7.0, it is highly recommended to use CSPro DB
instead. There are new features such as smart sync and case labels that are not supported using text files.

When you launch a CSPro data entry application you can select the data file name and the type of data file that you want to
use:

( B
HouseholdQuestionnaire Iﬁ

e CSPro DB: Thisis a CSPro Data Base File. It

Data File Name Type o contains the data and metadata.
Input Data File .. file name ... ... CSProDB ~
fi CSPoDB e Encrypted CSPro DB: A CSPro DB file with
';‘g;‘f encrypted data requiring a password to open.

¢ None: There is no associated data file. This is
useful when building menus.

e Text: Thisisa UTF-8 Text file. This is what
CSPro has used in previous versions and is
provided for compatability.

Controlling the size of the roster

Our program at this point has no way to stop entering household members unless we fill in all 50 rows of the roster. To limit
the number of rows in the roster to the number of actual household members we can have the interviewer enter the
number of household members and use that to set the size of the roster. Add a new variable, "number of household
members", to control the size of the roster. Which record should we add it to? It can't be on the person record since we
don't want it to repeat so let's add it to the household characteristics record but put it on the person (section B) form. Right
click on the roster, choose properties and set this new field as the occurrence control field.

Try running the application again. Unfortunately, when we get to the person form we go into the roster instead of the
"number of household members" field. By default, CSEntry goes in the order in which the fields were dropped on the form.
We can change the order by dragging the fields in the forms tree. Move the "number of household members" field up in the
tree so that it comes before the roster. Now the roster control field should work. In the next lesson we will use logic to exit
the roster without the roster control field.

Running the Application on Mobile
To put the application on a phone/tablet we need to do the following:

1. Publish the .pen file (File menu -> Publish Entry Application).

2. Connect the tablet to the PC using a USB cable. The tablet should show up like a USB drive.

3. Copy the files Popstan2020.pen and Popstan2020.pff into the CSEntry directory on the tablet.
4. Start CSEntry on the tablet.

5. Enter the data (note the differences between Mobile and Windows).

6. When you are done, connect the tablet back to the computer and copy the file Household.csdb from the tablet back
to the PC. One some tablets (Nexus 7 for example) the first time the data file is created it may not show up when
connected to the PC. If this is the case, reboot the tablet and it will.

When we copied the application to the tablet we copied the pen and the pff file. The pff file contains various parameters
about how to launch the data entry application including the data file to use. You can modify the pff file by right clicking on it
in Windows Explorer and choosing "Edit". This allows you to modify the name of the data file, force the application to start in
add or modify mode and to lock various parts of the user interface.

On Android, the list of available applications on the device is constructed by finding all the pff files in the CSEntry directory
and subdirectories so a pen file without a pff will not show up in the list.

Data Manager
Double-clicking on the file household.csdb opens the file in the Data Manager tool. Data Manager provides a simple way to
view the data that was entered.

Group exercise
Add a new record to the dictionary for section E of the questionnaire (deaths). Name this record "Deaths". How
many occurrences should it have? Don't include EO1 and EO2 in the new record as they do not have the same number
of occurrences as the other variables in this section. Instead, add them to the housing record. Create a new form for
section E and add the fields onto it to create a roster. Use E02 as an occurrence control field to the roster to limit the
number of rows to the number of deaths in the household. Test the application on both Windows and on Android.

Page 8 of 87 Session 01: Overview of CSPro, Dictionaryand Forms



Partial Save and Case Tree

It can be a pain when testing a question that is on the third form of a survey to have to reenter all of the data up to the
question you are testing. We can enable partial save under the data entry options so that we can exit data entry, modify the
application, and come back right to where we left off. While we are in the data entry options we can also enable the case
tree on both Windows and Android to make it easier to navigate around the questionnaire while we are testing. The case
tree is enabled by default on Android since Android only shows one question at a time but it is off by default on Windows.
Note that on mobile phones, since there is not enough space to display the case tree and the questions at the same time so
you need to tap on the big green "CS" in the top left corner of the screen to bring up the case tree.

In addition to the data entry options, there are other useful settings in the "Data Source Options" dialog. For example, here
you can enable automatic partial save at regular intervals. This is important for long interviews where if there is a problem
with the software or device part way through the interview you could lose data if the interview has not been partially saved.

Subitems

Let's add the Date of Birth (B06) to the application. In order to be able to look at both the date as an 8-digit number and
look at the day, month and year individually we can create an item with subitems. Subitems are items that are made up of a
subset of the digits of their parent item. Add the item for date of birth and the following subitems:

o vyear of birth (length 4)
e month of birth (length 2)

e day of birth (length 2)

We are putting year first then month and day because this format will work better with other CSPro features that we will
see later. Click on Layout in the toolbar to ensure that the item and subitem overlap. Add the subitems to the form, add the
value sets for each subitem and test the application. Note that when we add the subitems to the form we do not need keep
the same order that we have in the dictionary. On the form we can put the day, month then the year.

Linked Value Sets

Note that questions BO7 (place of birth) and B08 (residence 1 year ago) both have the same value set. We could simply
copy and paste the value set into both items, however, this would leave us with two copies of the same value set. If later on
we change one of them and forget to change the other, then they will be out of sync. This could cause consistency problems
later on. Instead, we can create the value set for BO7 and then paste it as a linked value set into B0O8. With a linked value
set, CSPro only stores one copy of the value set that is shared between both variables. This way if you edit the value set, the
change is reflected in both items.

Multiply Occurring Items

Let's add questions FO4 and FO5 on number of housing units to the application. FO4 is just a yes/no question. FO5 asks the
number of units of each of five different types of housing unit. We could make five different variables in the dictionary:
HOUSING_UNITS_ROUND_HUT, HOUSING_UNITS_DETACHED, HOUSING_UNITS_SEMI_DETACHED etc... To simplify our
dictionary, we can instead make a single variable in the dictionary, HOUSING_UNITS, with five occurrences: one for each
type of housing unit. When we drag this variable onto the housing characteristics form we will get a roster with 5 rows.

Occurrence Labels

With this approach, our form does not show the housing unit types but we can fix that by using occurrence labels. Select the
housing units variable in the dictionary and choose "Occurrence Labels..." from the Edit menu. Add the names of the five
types of housing units in the grid that comes up. Note that you can copy from Excel and paste into this dialog. Now when we
drag the variable to the form the roster shows the type of housing unit for each row.

Checkboxes

We could also use a multiply occurring item for question B11, disabilities, but that can be implemented more easily using
checkboxes. Checkboxes offer a friendly interface for multiple response questions by presenting a single screen with a
checkbox for each option rather than presenting the options one by one.

In CSPro multiple response questions are implemented as alpha variables whose length is the same as the number of options
that can be selected at the same time. The value set has a value for each option which is usually a single letter. The resulting
value is a string containing the values for each of the selected items.

For disabilities we will use the following value set:
Visual A
Hearing B
Speech C
Physical D
Mental E
Self-care F

If the interviewer checks the boxes for Visual, Physical and Mental the value for the variable will be "ADE". We will see later
how can convert the alpha value into a series of yes/no values to simplify analysis.

Capture Types
If you right click on the disabilities field on the form and choose "Field Properties..." you will see that the capture type is set
to check box. There are other capture types:

e Text Box: keyboard data entry

e Radio button: choose one option from many with radio button for each option

e Drop Down: choose one option from many without radio buttons

e Combo Box: combination of keyboard input and drop down (same as drop down on Windows)

e Check Box: choose multiple responses

When you drag an item on the form, CSPro sets the capture type based on the value set for that item. If there is no value set
when you drop the item, the capture type will be set to text box. You can always change the capture using the Field
Properties dialog.

Date Fields

Let's add the interview date to the identification section. Which record should it go on? It could go on any singly occurring
record such as housing but let's create a new record called INTERVIEW_REC to hold the section A items that are not part of
the ID items and add it there. We can add an eight-digit item for the interview date. We can also add subitems for the year,
month and day of the interview. Drag the date item onto the form. When dragging don't use the subitems, just the items.
Now change the capture type for the interview date to be Date and set the date format to be YYYYMMDD to match the
order of the subitems year, month and day.

Filling in Names First

Currently the interviewer has to fill in all the demographic information for the first person before moving on to the next
person in the household. In practice, it is simpler to have them fill in just the name, age, sex and relationship of all the
household members first and then fill in the remaining details only after all the basic information has been entered. To do
this we need to pull just the PERSON_NUMBER, NAME, SEX, RELATIONSHIP, AGE and DATE_OF_BIRTH into a separate
roster. Delete them from the existing DEMOGRAPHICS_ROSTER, create a new form before the DEMOGRAPHICS_FORM
called HOUSEHOLD_MEMBERS_FORM and drop PERSON_NUMBER onto it to make a new roster. Rename this roster
"HOUSEHOLD_MEMBERS_ROSTER". Drop NAME from the dictionary on top of the HOUSEHOLD _MEMBERS_ROSTER to
add it to the roster. Note that when you drop an item from a repeating record on top of an existing roster it gets added to
that roster but if you drop it outside the roster it creates a new roster. Drop the SEX, RELATIONSHIP, AGE and
DATE_OF_BIRTH variables onto the new roster. Remove the NUMBER_OF_HOUSEHOLD_MEMBERS field from the
DEMOGRAPHICS_FORM and drop it onto the HOUSEHOLD_MEMBERS_FORM. In the forms tree, move
NUMBER_OF_HOUSEHOLD members so that it comes before the HOUSEHOLD_MEMBERS_ROSTER. Set the occurrence
control field for HOUSEHOLD_MEMBERS_ROSTER to be NUMBER_OF_HOUSEHOLD_MEMBERS.

Tip: When dragging items onto a form with an existing roster drop the items inside the roster or you will end up with a
second roster for the new item.

Page 10 of 87 Session 01: Overview of CSPro, Dictionaryand Forms



Blocks

The blocks feature allows you to group several fields into a related unit. Blocks are primarily useful for two reasons. Firstly,
they provide the mechanism for CSEntry to display multiple fields on a screen. Secondly, they allow you to write logic checks
in one place that apply to a number of fields.

Let's add a block to group date fields. To create a new block, select day, month and year fields in the Forms Tree by holding
down the Ctrl key while selecting fields. Once the fields are selected, right-click and select Add Block. In the "Block
Properties" dialog box, type "Date of Birth" for the label and "DOB_BLOCK" for the name. Leave the checkbox option "Show
on same screen on mobile" checked to display the grouped fields on the same screen on mobile devices.

Exercises

1. Add the remaining fields from the housing section of the questionnaire to the dictionary and the housing form (FO6
through F13). Make sure to add the appropriate value sets.

2. Add the remaining fields from the demographics section (B) of the questionnaire to the dictionary and to the
demographics form. Make sure to add the appropriate value sets. For Language(B15) use checkboxes.

3. Add the fields for section C, Education, to the dictionary. Add them to the person record. Create a new form for
section C and drop the education items onto to it to create a roster. Set the occurrence control field for the roster to
the number of household members.

4. Addthe start and end times of the interview to the identification form after the interview date. Use subitems for the
hours and minutes. Add appropriate value sets.

5. Add a new record and form for section G, household possessions. Since GO1 (quantity and value) repeat, put these
items in their own repeating record but do not include GO02. This roster should include the possession code (1-10),
guantity and value per unit. Set the occurrence labels in the roster for GO1 to the names of the possessions. Make
GO02 a singly occurring checkbox field and put it in the housing record since it does not repeat. The value set for G02
should have the possession names as labels with codes "A", "B", "C"...

Make sure to test your application on both Windows and Android.

Session 02: Skips and Prefills

At the end of this session participants will be able to:

e Use the commands skip and ask 1if to skip fields
e Useif then else statementstoimplement skip patterns
e End rosters from logic using endgroup

e Automatically fill in fields using logic

Skips

In question FO4 (other housing units), we need to skip over question FO5 (numbers of housing units) if they answer "no".

To skip to a field, we need to use logic. To add logic to a CSPro application click on the logic button == on the toolbar to
switch to the logic view. Here we can type in CSPro logic to control the program. Logic is added to a PROC (procedure)
usually associated with a field on the form. To go to the PROC for a field in the logic view, just click on the field in the form
tree on the left while in logic view. We will add our skip to the PROC of field OTHER_HOUSING_UNITS so click on
OTHER_HOUSING_UNITS in the form tree.

To skip to a field, we use the command skip. In order to skip only when the answer is no, we need to combine the skip with
an if statement as follows:

PROC OTHER_HOUSING_UNITS

if OTHER_HOUSING_UNITS = 2 then
skip to TENURE;
endif

The if statement only executes the code between the then and the endif if the condition
(OTHER_HOUSING_UNITS = 2)istrue, otherwise it goes straight to the first command after the endif.

This will skip directly to TENURE for individuals who answer "no" to OTHER_HOUSING_UNITS, skipping over the
HOUSING_UNITS roster.

In addition to "=" to check if two values are equal, you can use the following operators:
Operation Symbol
Equal to =
Not equal to <>
Less than <
Less than or equal to <=
Greater than >

Greater thanorequalto  >=

For example, to skip over B08, residence one year ago, for those less than one year of age we could add the following logic
to the PROC of the preceding question:

PROC PLACE_OF_BIRTH
if AGE < 1 then

skip to MOTHER_ALIVE;
endif;

Note that once we have skipped the field RESIDENCE_ONE_YEAR_AGO, we can't get to it by clicking on it or going back
from the following field. CSEntry keeps track of the fact that this field was skipped and won't let us back in until we change
the value of AGE. This is an important difference between system controlled and operator controlled data entry mode.

We should add a comment to other readers of our logic so that they will understand what we are trying to do. This will not
only help others understand our logic but it will help us when come back to it after a few months.

PROC PLACE_OF_BIRTH
// Do not ask residence one year ago for those less than one year old
if AGE < 1 then
skip to MOTHER_ALIVE;
endif

Everything on a line after // is considered a comment and is ignored by CSPro. You can also use /* */ for multiline
comments. It is good practice to use plenty of comments to document your logic.

A note on coding style
When writing if statements, your code will be more readable and easier for others to understand if you indent the
statements between the then and the endif. It is also helpful to use consistent capitalization. We recommend all
uppercase for dictionary variables like NUMBER_OF_HOUSEHOLD_MEMBERS and all lowercase for CSPro keywords
like if, then and endif. Finally, you should use comments as much as possible to help others, and your future self,
better understand your code.

Ask if
An alternative to the skip is the statement ask if. It skips the current question if the condition is NOT met. For example, to

use ask ifinorderto only ask the question RESIDENCE_ONE_YEAR_AGO (B08) for those aged 1 and over we can use the
following logic:

PROC RESIDENCE_ONE_YEAR_AGO
preproc
ask if AGE >= 1

The ask if statement skips over the question unless the condition AGE >= 1 istrue. In other words, if the age is less than
1, the question will be skipped. Note that when changing from skip to ask if, the condition was negated by switching
from I|<|I tO |I>=II.

The line preproc tells CSPro to execute this logic before asking the question. Our previous examples did not specify
preproc, so by default the logic was executed in the postproc. Postproc logic is executed after the data are entered.
Preproc logic is executed before the data are entered. For each variable, you can specify a preproc, a postproc or both. The
ask if statement will always be used in the preproc of the field being skipped while skip can either be in the preproc of
the field being skipped or in the postproc of the preceding field.

Group exercise
Implement the skip pattern for questions F10 (has a toilet) and F11 (type of toilet). If the answer to question F10,
have toilet, is "no", skip question F11, type of toilet. You can use either skip or ask if. Make sure toadda
comment to your code and to format your code using correct indentation and capitalization.

Compound Conditions

As another example, we will implement the skip pattern for question C01. We need to skip over C02, highest level of
education, if school attendance is "never attended" (1) or "don't know" (9). We can implement this with two if statements:

PROC ATTENDED_SCHOOL

// Skip over highest level education for those who never attended school
if ATTENDED_SCHOOL = 1 then

skip to LITERATE;
endif;

// Skip over highest level education for those whose attendance is "don't know"
if ATTENDED_SCHOOL = 9 then

skip to LITERATE;
endif;

Note that in CSPro, logic statements must be separated by semicolons (;). This tells CSPro when one command ends and the
next begins. Forgetting to put the semicolon at the end of a statement is a very common error that new users make. If you
forget the semicolon, usually CSPro will tell you that that a semicolon was expected, although sometimes it gets confused
and gives you a less informative error message.

We can simplify this code by combining the two if statements using the "or" operator to create a compound condition.

PROC ATTENDED_SCHOOL

// Skip over highest level education for those whose attendance is "never attended"
// or "don't know"
if ATTENDED_SCHOOL = 1 or ATTENDED_SCHOOL = 9 then
skip to LITERATE;
endif;

In addition to the "or" operator you can also make compound conditions using the "and" operator. A compound statement
linked by "or" is true if either one of the conditions are true but a compound statement using "and" is only true if both
conditions are true.

In addition to "or", CSPro supports the following logical operators:

Operation Keyword
Negate an expression not
True if both expressions are true and

True if either expression is true or

You can also use compound conditions with ask if. For example we only want to ask question B14, age at first marriage,
for those who responded "married" or "widowed" or "divorced" for marital status.

PROC AGE_AT_FIRST_MARRIAGE

preproc

// Only ask age at first marriage for those whose marital status indicates
// they are/were married

ask if MARITAL_STATUS = 2 or MARITAL_STATUS = 3 or MARITAL_STATUS = 4;

An "or" expression is true when either the left or right expressions are true so the question will only be asked if the
MARITAL_STATUS is either 2, 3 or 4.

We can make our expression even simpler using the "in" operator which checks if a value is in a range:

PROC AGE_AT_FIRST_MARRIAGE

preproc

// Only ask age at first marriage for those whose marital status indicates
// they are/were married

ask if MARITAL_STATUS in 2:4;

Other (specify) fields

A common pattern in CSPro is to use the ask if statement to implement other (specify) fields. Let's implement the other
(specify) field for question FO8, roofing material. First we add a new alpha field to the dictionary to capture the "other"
value. We can call it ROOFING_MATERIAL_OTHER and drop it onto the form so that it comes right after
ROOFING_MATERIAL. We only want to ask for the "other" value if they selected "other" (code 5) for ROOFING_MATERIAL,
otherwise we want to skip it. We can do that using ask if in the preproc for ROOFING_MATERIAL_OTHER.

PROC ROOFING_MATERIAL_OTHER
preproc

// ask this question only if "Other (specify)" is picked in roofing material.
ask if ROOFING_MATERIAL = 5;

Skip to next

Now let's add the skip between sign language (B12) and marital status (B13) to skip to the next person for household
members under 10 years old. In this case, what do we skip to? We can't skip to BO1 (line number) since that will try to send
us backwards to the line number for the current row. Instead, we use skip to next which automatically skips to the first
field in the next row of the roster. We will put this skip in the postproc of SIGN_LANGUAGE.

PROC SIGN_LANGUAGE
// Skip to next person for household members under 10 years of age
if AGE < 10 then
skip to next;
endif;

Let's add the skip in section C, education, for those under 3 years of age. We are only supposed to fill in the education section
for those aged three years and above. We can use skip to next to skip over the education questions if the age is less
than three. We need to do this in the preproc of C01, the first field in the education roster, ATTENDED_SCHOOL.

Note that since this field already has a postproc we need to add the preproc first and then add the word "postproc™
so that the original postproc logic does not become part of the preproc.

PROC ATTENDED_SCHOOL
preproc

// Skip education for household members under 3 years of age
if AGE < 3 then

skip to next;
endif;

postproc

// Skip highest grade if never attended or don't know

if ATTENDED _SCHOOL = 1 or ATTENDED _SCHOOL = 9 then
skip to LITERATE;

endif;

Prefills
Let's fill in the line number automatically so that the interviewer doesn't have to enter it. We can do this in the preproc of
PERSON_NUMBER:

PROC PERSON_NUMBER
preproc

// Fill in line number automatically
PERSON_NUMBER = curocc();

The CSPro function curocc () gives us the current occurrence number of a repeating record or item. In other words, it
gives the row number of the roster we are currently on.

To avoid having to hit enter to move through this field we can use the statement noinput which accepts the assigned
answer and automatically moves to the next field as if the interviewer had used the enter key.

PROC PERSON_NUMBER
preproc

// Fill in line number automatically
PERSON_NUMBER = curocc();
noinput;

Alternatively, we can make the field uneditable by checking the protected box in the field properties. With noinput it is still
possible to go back and modify the field, but protected fields cannot be modified except from logic. Be aware that if you do
not set the value of a protected field in logic before the interviewer gets to the field, CSEntry will give you a fatal error.

Group Exercise
In question B10, mother line number, automatically fill in the code 88 (deceased) if the response to B09, mother
alive, is "no". You may be tempted to use the skip statement for this but you should use the noinput statement
instead.

Endgroup

Instead of using the occurrence control field in the roster, we could ask the user if they want to terminate the roster. To do
that, we first add a new field to the dictionary and it to the roster. Let's call it MORE_PEOPLE and give it the value set Yes -
1, No—2. We will put it at the end of the names roster. If the interviewer picks "no" then we use the command endgroup
which terminates entry of the current roster or repeating form.

PROC MORE_PEOPLE

// Exit roster when no more people in household
if MORE_PEOPLE = 2 then

endgroup;
endif;

With this, we no longer need to use the occurrence control field of the roster. However, since the other rosters
(demographics and education) still rely on the variable NUMBER_OF_HOUSEHOLD_MEMBERS as a roster control field we
need to set its value after completing the names roster. We can make it a protected field, move it after the names roster
and set it using logic:

PROC NUMBER_OF_HOUSEHOLD_MEMBERS

preproc

// Set number of household members to size of person roster.

// It is used as roster occurrence control field for later rosters.
NUMBER_OF_HOUSEHOLD_MEMBERS = totocc(HOUSEHOLD_MEMBERS_ROSTER);

This uses the function totocc () which gives the total occurrences of a repeating record or item. In other words, it gives us
the total number of rows of a roster.

Endlevel
There is also the endlevel command which is similar to endgroup but skips the rest of the current questionnaire (except
for two level applications when in a second level node where it terminates the current node).

Skipping Based on Checkboxes

We only want to ask question B12, sign language, if the household member is hearing disabled. We want to skip question
B12 if hearing disabled is not checked in question B11. How can we tell in logic if hearing disabled is checked? Hearing
disabled is option B in the value set so we need to determine if the set of all checked options contains the letter B.

In CSPro logic we can use the function ischecked () which returns true if a value is one of the selected values in a
checkbox variable. Using ischecked () the logic for skipping sign language if hearing is not checked is:

PROC SIGN_LANGUAGE

preproc
//ask this question only if hearing disabled (option B) is checked
ask if ischecked("B", DISABILITIES);

Note that since we are adding a preproc to a variable that already has a postproc we need to explicitly add the postproc.

PROC SIGN_LANGUAGE

preproc
//ask this question only if hearing disabled (option B) is checked
ask if ischecked("B", DISABILITIES);

postproc
// Skip to next person for household members under 10 years of age
if AGE < 10 then
skip to next;
endif;

While this works, it creates a problem with the skip to next inthe postproc. If we enter an individual with age less than
10 who has no difficulty hearing we no longer skip the marital status question. This is because the ask if inthe preproc
skips over the variable and the postproc is not run. Where can we put the skip to next sothat it will work for those who
have no difficulty hearing? We can move it to the preproc of MARITAL_STATUS.

PROC MARITAL_STATUS
preproc
// Skip to next person for household members under 10 years of age
if AGE < 10 then
skip to next;
endif;

Exercises

1. Skip question FO7, monthly rent, if dwelling is not rented in F06.
2. Insection E, deaths, implement the skip pattern for question EQ1, any deaths.

3. Insection E, deaths, implement the skip pattern for question E08, mother of deceased, based on the age at time of
death EO7.

4. Insection E, deaths, implement the skip pattern for question EQ9, died while pregnant.

5. Insection E, deaths, skip questions EQ9 and E10 for deceased household members who are NOT women aged 12-50.
6. Insection E, use logic to pre-fill the line number field E03. Make sure that the line number is protected.

7. Skip all of section C for persons under 3 years of age.

8. Add an additional question to F13 to determine if the respondent wants to give the distance to water in minutes or
kilometers. If they choose kilometers then skip the distance in minutes question otherwise skip the distance in km
question.

9. Add the other(specify) field and skip pattern for FO8, wall material.
10. Add the other(specify) field for B15, languages spoken.
11. Automatically fill in the the possession code in question GO1.
12. In question GO1, household possessions, skip the value field if the quantity is zero.

13. Add variables and a form for section D: Fertility. Add the variables to the person record but create a new form with
its own roster just like we did for section C. Don't worry about the last question on the form that displays the total
births. We will cover that in a later lesson. Add the skip patterns for the fertility section. Don't forget to skip the
entire section for males and females NOT between 12 and 50 years old.

Session 03: Consistency Checks

At the end of this session participants will be able to:

e Use the command errmsg to display messages to the interviewer

e Use the errmsg with select keyword

e Useif then else statementstoimplement consistency checks across multiple variables
e Use the command warning to implement "soft" checks

e Create and run test plans for consistency checks

e Implement consistency checks on dates

e Declare and use logic variables

Consistency Checks with Two Variables

Up to now, we have been able to limit the responses for each variable to only a set of valid responses using value sets. What
if we want to check for consistency between two different variables?

For example, let's prevent the interviewer from entering more bedrooms than there are rooms in the house. Click on the
NUMBER_OF_BEDROOMS field in the forms tree and then click on Logic in the toolbar to open the logic editor. Add the
following code in the procedure for NUMBER_OF_BEDROOMS:

PROC NUMBER_OF_BEDROOMS

// Ensure that number of bedrooms is not more than number of rooms.
if NUMBER_OF_BEDROOMS > NUMBER_OF_ROOMS then
errmsg("Number of bedrooms cannot exceed number of rooms");
reenter;
endif;

The if statement only executes the code between the then and the endif if the condition
(NUMBER_OF_BEDROOMS > NUMBER_OF_ROOMS) is true. The errmsg statement will display a message to the user. The
reenter statement forces the interviewer to stay in the current field and does not allow them to advance to the next field.

Note that we put our logic in the proc for NUMBER_OF_BEDROOMS and not in the proc for NUMBER_OF_ROOMS. This is
because when we are in the proc for NUMBER_OF_ROOMS, the value for NUMBER_OF_BEDROOMS has not yet been
entered. When creating consistency checks we always put the logic for the check in the proc for the last field involved in the
check.

Let's take a look at another example, the edit specifications call for the following control:

o |f RELATIONSHIP is spouse, then MARITAL_STATUS should not be coded 1 (never married), 2 (divorced) or 3
(widowed).

Which proc should we put this check in? Since MARITAL_STATUS comes after RELATIONSHIP we will put it in the proc for
MARITAL_STATUS.

PROC MARITAL_STATUS

// Ensure that spouse is not single, divorced or widowed
if (MARITAL_STATUS = 1 or MARITAL_STATUS = 3 or MARITAL_STATUS = 4)
and RELATIONSHIP = 2 then
errmsg("Marital status of spouse cannot be single, divorced or widowed");
reenter;
endif;

Note that when combining multiple "and" and "or" expressions, "and" expressions are evaluated first and then "or"
expressions are evaluated which can sometimes lead to unexpected results. You can use parentheses to force the order of
evaluation you want like we do above. What happens to the above expression without the parentheses if the relationship is
not 2 and the marital status is 3?

Of course there are a couple of simpler ways to write this check without using "or". We can use the not equals operator ("
<>|| :

// Ensure that spouse is not single

if MARITAL_STATUS <> 2 and RELATIONSHIP = 2 then
errmsg("Marital status of spouse cannot be single, divorced or widowed");
reenter;

endif;

Or we can use the keyword "in":

// Ensure that spouse is not single

if MARITAL_STATUS in 1,3,4 and RELATIONSHIP = 2 then
errmsg("Marital status of spouse cannot be single, divorced or widowed");
reenter;

endif;

The in operator will be true if the value matches any of the numbers in the comma separated list. It also supports ranges by
separating the start and end of the range with a colon (:). For example BEDROOMS in 1:4 will be true if BEDROOMS is
1,2,3 or 4.

Better Error Messages
We can improve our error message by adding parameters to the string. Let's go back to the check on the number of
bedrooms and display the number of rooms and the number of bedrooms in the message:

PROC NUMBER_OF_BEDROOMS

// Ensure that number of bedrooms does not exceed number of rooms.
if NUMBER_OF_BEDROOMS > NUMBER_OF _ROOMS then
errmsg("Number of bedrooms, %d, exceeds number of rooms, %d.",
NUMBER_OF _BEDROOMS, NUMBER_OF ROOMS);
reenter;
endif;

The "%d"s in the message are replaced by the values of the variables that follow in the order that they are listed.
Let's do the same for the check on marital status and relationship:

// Ensure that spouse is not single
if MARITAL_STATUS in 1,3,4 and RELATIONSHIP = 2 then
errmsg("Relationship is spouse and marital status is %d. Spouse cannot be
single, divorced or widowed.", MARITAL_STATUS);
reenter;
endif;

It would be better if we could include the name of the person in the error message as an additional clue to the interviewer.
Since the name is an alpha variable we use %s instead of %d. (%d is for decimal value.)

PROC MARITAL_STATUS

// Ensure that spouse is not single
if MARITAL_STATUS in 1,3,4 and RELATIONSHIP = 2 then
errmsg("%s has relationship spouse and marital status %d. Spouse cannot be
single, divorced or widowed.",
NAME, MARITAL_STATUS);
reenter;
endif;

Why is there a whole bunch of extra space after the name in the error message? Remember that alpha variables in the
dictionary are fixed length. This means that they get padded with blank spaces. We can remove the trailing blank spaces by
using the strip() function.

PROC MARITAL_STATUS

// Ensure that spouse is not single
if MARITAL_STATUS in 1,3,4 and RELATIONSHIP = 2 then
errmsg("%s has relationship spouse and marital status %d. Spouse cannot be
single, divorced or widowed.",
strip(NAME), MARITAL_STATUS);
reenter;
endif;

Finally, rather than show the numeric value of the marital status it would be nicer to show the label from the value set. We
can do this using %l instead of %d. This will display the label from the value set rather than the numeric code.

PROC MARITAL_STATUS

// Ensure that spouse is not single
if MARITAL_STATUS in 1,3,4 and RELATIONSHIP = 2 then
errmsg("%s has relationship spouse and marital status %1. Spouse cannot be
single, divorced or widowed.",
strip(NAME), MARITAL_STATUS);
reenter;
endif;

Errmsg with select

If we add select to the errmsg we can give the user the option of going back to correct either field. We can use this in
our bedrooms example:

PROC NUMBER_OF_BEDROOMS

// Ensure that number of bedrooms does not exceed number of rooms.
if NUMBER_OF BEDROOMS > NUMBER _OF ROOMS then
errmsg("Number of bedrooms, %d, exceeds number of rooms, %d.",
NUMBER_OF_BEDROOMS, NUMBER_OF_ROOMS)
select("Fix number of rooms", NUMBER_OF_ROOMS,
"Fix number of bedrooms", NUMBER_OF_BEDROOMS);
endif;

With the select we no longer need the reenter since CSEntry will automatically reenter the field that the interviewer
selects.

Note that the select clause is part of the errmsg statement so there is no semicolon in between the select and the
errmsg.

Let's add a select to the relationship-marital status check too:
// Ensure that spouse is not single
if MARITAL_STATUS in 1,3,4 and RELATIONSHIP = 2 then
errmsg("%s has relationship spouse and marital status %1. Spouse cannot be
single, divorced or widowed.",
strip(NAME), MARITAL_STATUS)
select("Fix marital status", MARITAL_STATUS,
"Fix relationship”, RELATIONSHIP);
endif;

Soft edit checks
If you want to allow the user to ignore the error and enter the next field, you can add an option to the select statement with
the keyword continue:

PROC NUMBER_OF_BEDROOMS

// Ensure that number of bedrooms does not exceed number of rooms.
if NUMBER_OF_BEDROOMS > NUMBER_OF ROOMS then
errmsg("The number of bedrooms, %d, is greater than the number of rooms, %d.",
NUMBER_OF BEDROOMS, NUMBER_OF ROOMS)
select("Fix number of rooms", ROOMS,
"Fix number of bedrooms", NUMBER_OF_BEDROOMS,
"Ignore error", continue);
endif;

Now the message dialog will have a third button labelled "ignore" that, when clicked, will move to the next field, in this case
to type of main dwelling.

This is commonly referred to as a "soft edit check" as opposed to the previous "hard edit check" that does not allow the
interviewer to move on until the inconsistency is fixed. The advantage of a soft edit check is that the interviewer won't get
stuck, however, data quality may suffer. In a CAPI census soft edit checks are generally preferred in order to prevent
interviewers from getting stuck and to keep interview times to a minimum.

For soft edit checks, CSPro provides a warning function, which is identical to errmsg except that if you have already ignored
the message once, when you navigate through the field by clicking on the case tree or resuming from partial save, the
message is not shown a second time. To make the rooms/bedrooms a soft check, in addition to adding the ignore option, we
would use warning instead of errmsg.

PROC NUMBER_OF_BEDROOMS

// Ensure that number of bedrooms does not exceed number of rooms.
if NUMBER_OF BEDROOMS > NUMBER _OF ROOMS then
warning("The number of bedrooms, %d, is greater than the number of rooms, %d.",
NUMBER_OF_BEDROOMS, NUMBER_OF_ROOMS)
select("Fix number of rooms", ROOMS,
"Fix number of bedrooms", NUMBER_OF_BEDROOMS,
"Ignore error", continue);
endif;

Testing Consistency Checks

When testing consistency checks involving multiple variables it is important to test all possible combinations of the variables.
To do this we can create a test matrix. For example, to test all possible combinations of our consistency check between
MARITAL_STATUS and RELATIONSHIP we would use the following matrix that shows the expected result for each
combination of the variables involved:

Marital Status

Relationship Never married (1) Married (2) Divorced (3)  Widowed (4)
Spouse (2) Error OK Error Error
Not-spouse (<> 2) OK oK OK OK

Using this matrix, you can test each of the 8 possible combinations and make sure that you get the expected result. This
ensures that all possible cases are tested.

By combining the test matrices from all the consistency checks in the application, you can create a test plan for the
survey/census application to verify the application works correctly when changes are made. Often times changes to one part
of the questionnaire can have unintended effects on skips and consistency checks in other parts of the questionnaire so it is
important to have a test plan that is used after every set of changes to ensure that no problems are introduced.

Calculations
In question D02 the respondent gives the number of boys living in the household, girls living in the household and the total
number of children living in the household. Let's check that the sum of the girls and the boys is equal to the total.

if BOYS_IN_HOUSEHOLD + GIRLS_IN_HOUSEHOLD <> CHILDREN_IN_HOUSEHOLD then
errmsg("Total number of children (%d) must equal the sum of the number of boys
(%d) and the number of girls (%d).",
CHILDREN_IN_HOUSEHOLD, BOYS_IN_HOUSEHOLD, GIRLS_IN_HOUSEHOLD)
select("Correct total", CHILDREN_IN_HOUSEHOLD,

"Correct boys", BOYS_IN_HOUSEHOLD,

"Correct girls", GIRLS_IN_HOUSEHOLD);
endif;

CSPro logic supports the following mathematical operators:

Addition +
Subtraction -
Multiplication *
Division /

Modulo (remainder) %

Exponentiation A

Checking Dates

Let's add a consistency check between the date of birth and the age. We can use the function datedif+ () to calculate the
age based on the interview date and the date of birth:

What proc should we put this check in? We need the age and the entire date of birth (year, month and day) to be entered
before we can do the check. We could therefore do this check in the postproc of the last of the date components or we could
put it in the postproc of the block that contains all of the date fields. The postproc of the block is executed after all the fields
in the block have been entered. On mobile, when the block fields are displayed on the same screen, the procs of the fields in
the block are not called until the user completes the entire block and moves to the next page. To ensure consistent behavior
between Windows and Android it is therefore recommended to put all validation code in the postproc of the block.

PROC DOB_BLOCK

// Ensure that age matches date of birth
if datediff(DATE_OF _BIRTH, DATE_OF_ INTERVIEW, "y") <> AGE then
errmsg("Age (%d) does not match date of birth (%d)", AGE, DATE_OF_BIRTH)
select("Correct Age", AGE, "Correct date of birth", DATE_OF BIRTH);
endif;

This works but it would be better if the error message informed the interviewer what the date of birth that we calculated
was.

PROC DOB_BLOCK

// Ensure that age matches date of birth
if datediff(DATE_OF BIRTH, DATE_OF INTERVIEW, "y") <> AGE then
errmsg("Age (%d) doesn't match date of birth (%d). Based on date of birth age
should be %d",
AGE, DATE_OF BIRTH, datediff(DATE_OF_BIRTH,
DATE_OF_INTERVIEW, "y"))
select("Correct Age", AGE, "Correct date of birth", YEAR_OF_BIRTH);
endif;

We should also handle the case where the age or date of birth is unknown.

PROC DOB_BLOCK

// Ensure that age matches date of birth
if AGE <> 999 and DATE_OF BIRTH <> 99999999 then
if datediff(DATE_OF_BIRTH, DATE_OF_INTERVIEW, "y") <> AGE then
errmsg("Age (%d) doesn't match date of birth (%d). Based on date of birth
age should be %d",
AGE, DATE_OF BIRTH,
datediff(DATE_OF BIRTH, DATE_OF INTERVIEW, "y"))
select("Correct Age", AGE, "Correct date of birth",
YEAR_OF_BIRTH);
endif;
endif;

Making sure we have the correct number of 9s is a bit tricky. Since we used the special value missing for 999 in the value set
for age we can instead use missing in our if statement. We can do the same for the date of birth.

// Ensure that age matches date of birth
if AGE <> missing and DATE_OF_BIRTH <> missing then
if datediff(DATE_OF BIRTH, DATE_OF INTERVIEW, "y") <> AGE then
errmsg("Age (%d) doesn't match date of birth (%d). Based on date of birth
age should be %d",
AGE, DATE_OF_BIRTH,
datediff(DATE_OF_BIRTH, DATE_OF_INTERVIEW, "y"))
select("Correct Age", AGE, "Correct date of birth",
YEAR_OF _BIRTH);
endif;
endif;

This not only makes our code clearer but is also more resilient since the logic would still work if we later changed the size of
the variable AGE from three digits to two.

Logic variables

It would better if we didn't repeat the datediff calculation twice. Repeating code makes it harder to maintain. We may fix a
bug in one copy of the code but forget to do so another. To avoid repeating ourselves we can declare a logic variable to hold
the value of our calculation. Logic variables are like dictionary variables but are only for use in calculations in logic and don't
get shown on forms or saved to the data file. To create a variable, we declare it as follows:

numeric aNumber;
string anAlphanumeric;
alpha(20) anAlphaNumericWithAFixedLength;

Numeric variables hold numbers (including numbers with fractional parts) and string variables hold alphanumeric values.
Alpha variables are like strings but with a fixed length. In early versions of CSPro string variables didn't exist and it was
common to use alpha variables. In modern CSPro you should always use string variables instead of fixed length alpha
variables.

Coding style
We recommend using mixed case (a.k.a. camelCase) for logic variables in order to distinguish them from dictionary
variables.

We can declare a variable to hold the result of the datediff as follows:

numeric calculatedAge;
calculatedAge = datediff(DATE_OF_BIRTH, DATE_OF_INTERVIEW, "y");

Now we can use that variable in place of the datediff:

if calculatedAge <> AGE then
errmsg("Age (%d) does not match date of birth (%d). Based on date of birth age
should be %d",
AGE, DATE_OF BIRTH, calculatedAge)
select("Correct Age", AGE, "Correct date of birth", YEAR_OF_BIRTH);
endif;

It is possible to combine the declaration and the initialization of the variable into a single line:

numeric calculatedAge = datediff(DATE_OF_BIRTH, DATE_OF_INTERVIEW, "y");

When to Use Consistency Checks

It is tempting to place consistency checks on every question in your application in order to maximize data quality but it is
important to realize that every check you add comes at a price. Beyond the time it takes to implement and properly test
each check, you also need to consider the additional time required to train and support a more complex application. Every
check you add increases the risk that there will be bugs in your application when it gets to the field. Unlike a key from paper
application, a bug in the field can mean an enumerator being stuck and abandoning an entire questionnaire. For surveys, you
generally have a small sample and better trained enumerators so this is less of a concern. However, for a census,
enumerators are not as experienced and supporting them in the field is much more of a challenge. In a census, due to the
number of respondents, you can use imputation to correct many inconsistencies with minimal impact on the overall results.
In any case, just the fact that the CAPI instrument imposes range checks and skip patterns will be a significant improvement
in data quality over a paper census. For a CAPI census application, we recommend only adding consistency checks on the key
demographic fields in the household roster and leaving potential inconsistencies in other parts of the questionnaire to be
corrected post data collection. In addition, we recommend using mainly "soft" edit checks so that the enumerator is never
blocked by the application.

Exercises
Implement the following consistency checks from the edit specifications. Test that they are correct. Make sure to test all
possible cases (test matrices can help with this).

1. Display an error message if the head of household is less than 12 years old.
2. Ensure that B14 (age at first marriage) is less than or equal to the current age (B05).

3. Display an error message if the highest level of education (€C02) is university or graduate but the age is less than
eighteen. Make this a "soft check" so that the interviewer can ignore the error if they wish.

4. Add an error message if the highest level of education (€C02) is "standard 5" or higher and the individual does not
know how to read or write (C03). Make this a soft check.

5. Add a consistency check in section F that displays an error message if the household has a flush toilet in question F11
and does NOT also specify piped water inside house for question F12. Use a select clause to allow them to correct
either F11 or F12.

6. Ensure that the total number of children in D03 is equal to the sum of the boys and the girls. Do the same for question
DO4.

7. Check that the date that the death occurred in question E05 is within the last five years, since the section should only
include deaths occurring in that period. Use the interview date to calculate the number of years since the death
occurred. For example, if the interview date is 2016-10-05 and the date of death is 2016-09 then you should show an
error message. You can use datediff to calculate the difference between the interview date and the date of death as
we did for the age and date of birth check, however you will need to construct a full date from just the month and
year of death. Since no day is given, use 1 as the day (assume first day of the month). Note that datediff rounds to the
nearest year so to do the check correctly you will need to get the difference in months and compare it to 60 (5 years
times 12 months).

Session 04: More Logic in Data Entry, Rosters,
Subscripts and Loops

At the end of this session participants will be able to:

e Implement consistency checks across individuals in the household roster

e Understand the order of PROCs and where to place edit checks

e Use subscripts, totocc (), count(), seek() and curocc() to work with rosters

e Useloops(do, while)toimplement edit checks on repeating records/items (rosters)

e Understand and use "special values" (notappl, missing, refused, and default)in logic

e Set occurrence labels on rosters from logic

Groups and Subscripts

Let's add a check to the line number of mother to ensure that the line number entered corresponds to a woman 12 or over
in the household. To do this we need to link the line number entered in B10 back to the corresponding row of the roster in
section B. This is done through subscripts to get the age and sex of the mother based on the line number.

For any item that is repeated, adding a number in parentheses after it gives you the value of a particular occurrence of the
variable. For example, AGE (1) will be the age of the first household member, AGE (2) the age of the second member... If
we omit the subscript when executing logic in a proc of a roster CSPro will assume that we want the one for the current row
of the current roster. However, in this case we don't want the sex and age of the person in the current row, we want the
age and sex in the row specified by MOTHER_LINE_NUMBER. To get that we need to use AGE (MOTHER_LINE_NUMBER)
and SEX(MOTHER_LINE_NUMBER).

PROC MOTHER_LINE_NUMBER

// Ensure that line number of mother is line number of woman over 12
if not MOTHER_LINE_NUMBER in 87:88 then

if SEX(MOTHER_LINE_NUMBER) = 1 then
errmsg("Sex of mother %s is male",
strip(NAME (MOTHER_LINE_NUMBER)))
select("Correct mother",
MOTHER_LINE_NUMBER,
"Correct sex of " +
strip(NAME (MOTHER_LINE_NUMBER)),
SEX(MOTHER_LINE_NUMBER));
endif;

if AGE(MOTHER_LINE_NUMBER) <> 999 and AGE(MOTHER_LINE_NUMBER) < 12 then
errmsg("Age of mother %s is %d but must be at least 12",
strip(NAME(MOTHER_LINE_NUMBER)),
AGE (MOTHER_LINE_NUMBER))
select("Correct mother", MOTHER_LINE_NUMBER,
"Correct age of " + strip(NAME(MOTHER_LINE_NUMBER)),
AGE (MOTHER_LINE_NUMBER));
endif;
endif;

Blanks and special values

What happens if we refer to an occurrence of a row in the section B roster that doesn't exist? Try entering a line number for
the mother greater than the number of rows in the section B roster. Our tests for age and sex are not triggered. What are
the values of NAME, SEX and AGE when they are empty? Let's print them out using errmsg and see.

errmsg("NAME = %s, AGE=%d, SEX=%d", strip(NAME(MOTHER_LINE_NUMBER)),
AGE (MOTHER_LINE_NUMBER), SEX(MOTHER_LINE_NUMBER));

The alphanumeric variable NAME is just empty but the numeric values AGE and SEX are notappl. What does this mean?
Generally, numeric fields that are blank (skipped or not yet entered) have a special value called hotapp1l that can be used
in comparisons in logic. For example, to test if the SEX is blank we can use the following comparison:

if SEX(MOTHER_LINE_NUMBER) = notappl then
// Sex is blank, must be an empty row in section B roster
errmsg("%d is not a valid line in section B", MOTHER_LINE_NUMBER);
reenter;

endif;

There are other special values that are used in CSPro:

e missing: can be used as an alias for no response codes (9, 99, 999...). You must create an entry for it in the value
set.

e refused: similar to missing but can add special handling to make it more difficult to select. You must create an
entry for it in the value set.

e default: results from an error reading from a data file or from a calculation error (like trying to calculate notappl +
2 or dividing by zero) or moving a value into a variable that is too small to hold the value. For example, if AGE is a two
digit field, AGE = 118 will result in default.

Getting the size of a roster
In our error message, it would be nice to tell the interviewer what the maximum valid line number is. We can do that using
the function totocc () which gives you the total number of occurrences of a group.

if MOTHER_LINE_NUMBER > totocc(HOUSEHOLD_MEMBERS_ROSTER) then
// This is beyond the end of the roster
errmsg("%d is not a valid line in section B. Must be between 1 and %d.",
MOTHER_LINE_NUMBER, totocc(HOUSEHOLD_MEMBERS_ROSTER));
reenter;
endif;

Setting Occurrence Labels in Logic

If you look at the case tree in Android you will see that the roster occurrences are displayed with the name of the roster and
the occurrence number "Demographics(1), Demographics(2)..." which is not useful. Using logic, we can set the occurrence
labels to the names of the individuals instead. For that we use the command setocclabel () which takes the name of the
group (roster or repeating form) and the string to set it to. For example, to set the occurrence label of each row of the
different rosters once the name is entered we can do the following in the postproc of the name field (NAME):

PROC NAME

setocclabel (HOUSEHOLD MEMBERS ROSTER(curocc()), strip(NAME));
setocclabel (DEMOGRAPHICS ROSTER(curocc()), strip(NAME));
setocclabel (EDUCATION_ROSTER(curocc()), strip(NAME));
setocclabel (FERTILITY_ROSTER(curocc()), strip(NAME));

This works fine when we are adding a new case, however we open an existing case in modify mode the occurrence labels are
not set until we get to the person roster even though the occurrences already exist. In modify mode, while still on the
demographics roster you can scroll the case tree to see Demograpbhics (1), Demographics (2)... as we had before. In order to
prevent this, we need to set the occurrence labels for the rosters as soon as we open the case.

More About Procs

What proc can we use to set the occurrence labels? We could use the preproc of the first field of the first form but there is a
better option. It turns out that every element in the form tree has a PROC. Not only do variables have procs but there are
procs for forms, rosters, levels and even the application itself. Everything you see in the forms tree can have both a preproc
and a postproc. Understanding the order in which procs are executed is important in understanding CSPro logic.

The general rule is that

1. Parent items have their preproc called first,
2. then the procs of the child items are called,

3. and finally, the postproc of the parent is called.

Group Exercise: Proc Order
Form teams of three to five people. The instructor adds errmsg to the postproc and preproc of the following:
POPSTAN2020_FF (application), POPSTAN2020_QUEST (level), HOUSEHOLD_MEMBERS_FORM (form),
HOUSEHOLD_MEMBERS_ROSTER (roster), NAME (variable), NUMBER_OF_ROOMS (variable). Each team receives
slips of paper with the names of each preproc and postproc. BEFORE running the application, the teams have to put
the slips in the order that the errmsgs will be shown when the application is run. Teams have three minutes to
complete the exercise. Then the application is run and teams see if they have the correct results.

Setting Occurrence Labels (take 2)
Now that we understand procs, which proc should we set the occurrence labels in? The preproc of the questionnaire! We
can do something like the following:

PROC POPSTAN20208_QUEST
preproc

setocclabel (HOUSEHOLD MEMBERS_ROSTER(1), strip(NAME(1)));
setocclabel (HOUSEHOLD MEMBERS_ROSTER(2), strip(NAME(2)));
setocclabel (HOUSEHOLD MEMBERS_ROSTER(3), strip(NAME(3)));

Loops

The problem is that we want to do this for each household member, but the number of household members varies from case
to case. The above works only if we know the size of the household in advance. In order to handle any size household, we
need a loop. We can use a do loop which lets you repeat something until a certain condition is true. How many times do we
loop? We use the function totocc () that gives us the total number of occurrences of the roster.

PROC POPSTAN2020_ QUEST
preproc

// Fill in occurrence labels in rosters when entering a case that

// has existing data (partial save or modify mode). If we don't do
// this then the case tree will not have correct occurrence labels
// until after we pass through the household members roster.

do numeric i = 1 while i <= totocc(HOUSEHOLD_ MEMBERS_ROSTER)
setocclabel (HOUSEHOLD MEMBERS_ROSTER(i), strip(NAME(i)));
setocclabel (DEMOGRAPHICS ROSTER(i), strip(NAME(i)));
setocclabel (EDUCATION ROSTER(i), strip(NAME(i)));
setocclabel (FERTILITY_ROSTER(i), strip(NAME(i)));

enddo;

Counting Heads of Household

It is currently possible to enter multiple heads of household or no head of household at all. How can we add a consistency
check to ensure that there is exactly one head of household? In which proc would such a check go? We can put it in the
postproc of the roster since that will be run after all the rows have been entered.

How do we determine the number of heads of household? In each row of the roster, we can check the relationship to see if it
is head of household. To do this we need to a logic variable to keep track of the number of heads of household.

numeric numberOfHeads = 0;
Now we can increment the variable for each head of household we find:

PROC HOUSEHOLD_MEMBERS_ROSTER

// After person roster is complete, check to make sure that there is
// exactly one head of household.

numeric numberOfHeads = 0;

if RELATIONSHIP(1) = 1 then
numberOfHeads = numberOfHeads + 1;
endif;

if RELATIONSHIP(2) = 1 then
numberOfHeads = numberOfHeads + 1;
endif;

if RELATIONSHIP(3) = 1 then
numberOfHeads = numberOfHeads + 1;
endif;

if numberOfHeads <> 1 then
errmsg("Number of heads of household must be exactly one");
reenter RELATIONSHIP(1);

endif;

Since there is more than one row in the roster we need to use a subscript to tell CSEntry which row of the roster to look in.
RELATIONSHIP(3) refersto the relationship of the 3rd row of the roster.

The above works but only if we know the size of the household in advance. In order to handle any size household, we need a
loop. We can use a do loop. Just like with the occurrence labels, we can loop from 1 to the number of rows in the roster.

PROC HOUSEHOLD_MEMBERS_ROSTER

// After person roster is complete, check to make sure that there is
// exactly one head of household.

numeric numberOfHeads = 0;
do numeric i = 1 while i <= totocc(HOUSEHOLD MEMBERS_ ROSTER)
if RELATIONSHIP(i) = 1 then
numberOfHeads = numberOfHeads + 1;
endif;
enddo;

if numberOfHeads <> 1 then
errmsg("Number of heads of household must be exactly one");
reenter RELATIONSHIP(1);

endif;

Activity: Acting out a loop
Show a household with four members on the whiteboard/flipchart and show the code above on the screen. Pick
three volunteers. One volunteer plays the role of the variable numberOfHeads, a second plays the role of the variable
i and a third is the loop director. Give them each a sign so everyone knows who they are playing. The two people
playing variables should show their current values by holding up the correct number of fingers. The director is
responsible for walking through the code line by line and updating the variable values until the loop ends explaining
what they do step by step.

Now that we all understand loops let's see how we can simplify our code by using the function count () instead of a loop.

PROC HOUSEHOLD_MEMBERS_ROSTER

// After demographics roster is complete, check to make sure that there is
// exactly one head of household.

numeric numberOfHeads = count(PERSON_REC where RELATIONSHIP = 1);

if numberOfHeads <> 1 then
errmsg("Number of heads of household must be exactly one");
reenter RELATIONSHIP(1);

endif;

Force Head of Household to First Row
Instead of allowing the interviewer to enter the head of household on any line of the roster, we can greatly simplify our logic
if we force them to list the head on the first row. We can implement this in the postproc of RELATIONSHIP.

PROC RELATIONSHIP
// Don't allow non-head of household on first row
if RELATIONSHIP <> 1 and curocc() = 1 then
errmsg("Please enter the head of household before the other household
members.");
reenter;
endif;

// Don't allow head of household on a row other than first row
if RELATIONSHIP = 1 and curocc() <> 1 then
errmsg("0Only one head of household is allowed. Head of household was already
entered.");
reenter;
endif;

This replaces our logic for counting the number of heads of household in the postproc of the roster. This is simpler and will
also simplify any other checks involving the head. For example to ensure that the head of household is at least 12 years older
than his/her children we can simply compare the age of the individual to the age of head. Since the head is on the first row,
the age of the head will always be AGE (1). Since the head is entered first, we can do this type check in the proc for AGE
instead of in the roster postproc.

PROC AGE
// Compare age of child to age of the head
if RELATIONSHIP = 3 and AGE(1) - AGE < 12 then
errmsg("Child %s is %d years old but head %s is %d. Parent must be at least 12
years older than child.",
strip(NAME),
AGE,
strip(NAME(1)),
AGE (1))
select("Correct age of " + strip(NAME), AGE,
"Correct age of " + strip(NAME(1)), AGE(1),
"Correct relationship of " + strip(NAME),
RELATIONSHIP);
endif;
enddo;

Seek

Question D02 in the fertility section asks if a women in the household has any children that are living with her in the
household. If she answers "yes", then there should be at least one household member in who has listed her as their mother
by specifying her line number for B10, mother line number. We can use the seek function to find the line number of a child
whose mother line number matches the line number of the woman on the current line. If no such child exists then seek will
return zero.

PROC HAVE_CHILDREN_IN_HOUSEHOLD

// Check household roster to make sure that number of children
// in household roster based on mother line number is consistent
// with this answer.
numeric womanLineNumber = curocc();
numeric childLineNumber seek (MOTHER_LINE_NUMBER = womanLineNumber);
if childLineNumber > © and HAVE_CHILDREN_IN_HOUSEHOLD = 2 then
warning("There are children in the household who have listed %s as their
mother", strip(NAME))
select("Correct have children in household", HAVE_CHILDREN_IN HOUSEHOLD,
"Correct mother line number", MOTHER_LINE_NUMBER(childLineNumber),
"Ignore error", continue);
elseif childLineNumber = © and HAVE_CHILDREN_IN_HOUSEHOLD = 1 then
warning("There are no children in the household who have listed %s as their
mother", strip(NAME))
select("Correct have children in household", HAVE_CHILDREN_IN_HOUSEHOLD,
"Correct mother line number", MOTHER_LINE_NUMBER(1),
"Ignore error", continue);

endif;

Exercises

1. Ensure that the age difference between the head of household and the grandchild should not be less than 24 years. If
age of head is X years and the age of the child is Y years, then X-Y > 24.

2. Show an error message if the head of household has at least one spouse in the household and the marital of status of
the head of household is not married.

3. Show an error message if the head of household and his/her spouse are of the same gender. Make sure that your
logic works with polygamous households as well as households with just one spouse.

4. Set the occurrence labels in the section E roster, deaths, to the names of the deceased, E04, in that section. Make
sure that the labels are set correctly when resuming from partial save.

Session 05: CAPI Features

At the end of this session participants will be able to:

e Add question text and help text

o Add fills in question text

e Use multiple languages in question text, error messages and the dictionary
e Use occurrence labels as fills in question text

e Use setcaselabel to customize the case listing

e Use dynamic value sets

Question Text

We can add text to each question in our survey by clicking on "CAPI Questions" in the toolbar. This allows us to enter literal
question text for the questionnaire that the interviewer will read verbatim. Additionally, we can add interviewer
instructions. Let's start with the first few questions in section B.

For RELATIONSHIP, enter "What is (name's) relationship to the head of the household?" For SEX enter "Is (hame) a male or
a female?". Run the application on Windows and then on mobile to see how the question text is displayed.

In addition to the text, we can add instructions to the interviewer. For example for AGE we can have:
How old is (name)?

Followed by the instruction:
Enter age in completed years ("0" for children less than one year old)

To make it clear to the interviewer that this is an instruction, we use italics to distinguish it from the question. We can also
use different colors and fonts as well. You can use whatever scheme you like as long as it is consistent throughout the
application.

If you have your question text in Word or Excel you can copy and paste into CSPro and it will preserve the formatting.

Help Text

In addition to question text, we can provide additional instructions to the interviewer as "help text". Help text is not shown
by default but can be displayed by using the F2 key on Windows or tapping the help icon next to the question text on Mobile.
Let's add the following help text to the NAME field in section B:

Include all persons living in this house who have common arrangements for cooking and dining.

Run the application on both Android and Windows and see how the help text is displayed.

Fills in Question Text

It is possible to have the question include the values of dictionary variables. For example, for SEX, instead of asking "Is
(name) a male or a female?" we can include the respondents name by using ~~*NAME~" inside the question text. At runtime,
this will be replaced by the contents of the variable NAME.

Let's change the text for RELATIONSHIP, SEX and AGE to use ~~*NAME~~ in place of "(name)".

Conditional Question Text
Sometimes, the question text will be completely different depending on other factors and using fill variables is not sufficient
to implement the differences. For example, the question text for the NAME field should really be:

What is the name of the head household? — for the first row of the roster

What is the name of the next member of the household? — for the remaining rows of the roster.

We can implement this using the conditions window underneath the question text window. By default, this window contains
a single empty condition. Right-clicking on the conditions lets you add a new condition. Let's modify the first condition to be
curocc() = 1andthenadda second condition curocc() > 1. Now we have two different question texts, one that
will be shown for the first occurrence and the other that will be shown for the rest. Edit these two questions texts as above
and test it to verify that it works.

r

Condition

curocc)>1

curocc() = 1

It is possible to specify any logic expression in the condition column to determine whether or not to display the text. For
example setting the condition to "SEX=1" will display the question text only for males.

Supporting Multiple Languages
It is possible to add question text, dictionary labels and error messages in multiple languages.

Multiple Languages for Question Text

To have multiple languages for question text you first need to define the languages. This is done from the CAPI Options
Menu. Select "Define CAPI Languages". This will bring up the "Language" dialog box. Click on "Add" to add a language,
specify the name and label to use for the language and click "OK". The label is what the interviewer will see when choosing a
language and the name is what will be used in logic to refer to the language. The name should be a two letter abbreviation
for the language from the 1SO 639-1 standard such as EN, FR and ES. While you may use any language name, if you use one
of the ISO 639-1 codes then CSPro will automatically set the language of your application to the language set in the system
settings of your mobile device.

Languages >
Add | Remove | Maodify |
Hame Labal
gev  Engish
2 FR Francais
oK Cancel Help

When creating question text, you enter the question text for each language specified. Let's look at the question text for the
"Sex" item. There is a Drop Down menu for the languages we specified in the above steps. We select the language and then
enter the question text in that language.

For example, to add French translations for the sex, relationship and age questions we would enter:

Est-ce que ~~NAME™~~ est masculin ou feminin?

Quel est le lien de parenté de “~NAME~"~ avec le chef de ménage?
Quel est I'age de ~~*NAME~~?
Entrez I'dge en années révolues (entrez 0 pour les enfants dgés de moins de 1 an). Si inconnu entrez 999.

Now that you have created the question text in the specified languages, they can be selected during data entry.

In Windows, we can go to the "Options" Menu and select "Change Language" and a menu of the languages we defined will
be displayed. Click on the desired language and the question text will be displayed in that language. On mobile devices you
will find "Change Language" on the menu during data entry.

Multiple Languages for Dictionary Items

To have multiple language in value sets, you first need to define the languages. This is similar to the process we did to define
multiple languages for question text; however, for dictionary items you define languages from the Edit menu of the
dictionary. Make sure you are in the dictionary editor (you may need to click on the little blue book in the toolbar) and
choose "Edit" and then "Languages". This dialog is identical to the one used for adding languages to question text. You
should make sure to use the same names and labels for languages in this dialog as you entered for question text.

Once you have added additional languages to the dictionary you will see a language dropdown on the toolbar. To add a
translation for a dictionary label or value set label, choose the language in the dropdown and edit the label as you normally
would. This will set the label for the language you chose.

Multiple Language Error Messages

To add translations of text specified in program logic, such as arguments to errmsg( ), we can provide translations in the

message file. To do this we add the messages to the messages tab at the bottom of the logic editor. Instead of putting the

message text in the logic we tag each message with a number and use that number in the logic. In the message tab we list
the message numbers followed by the message text. This allows us to supply multiple versions of the message; one for each
language. Mark different languages in the file using "Language=" followed by the language name.

/* Application 'POPSTAN2020' message file generated by CSPro */
Language=EN

100 Head of household must be at least 15 years old

101 Correct age

102 Correct relationship

Language=FR

100 Le chef de menage doit avoir au moins 15 ans.
101 Corriger 1l'age

102 Corriger le lien de parente

In the logic you can pass the message number directly to errmsg (). Inside the select clause, however, you will need to use
tr() or maketext () around the message number.

PROC AGE

// Ensure that head of household is at least 15 years old
if AGE < 15 and RELATIONSHIP = 1 then
errmsg(100)
select(tr(101), AGE, tr(102), RELATIONSHIP);
endif;

Now when we run the application with the language set to French and encounter the error, it will show the French text.

You can also use a separate message file for each language. To add an additional message file, use "Add files" from the files
menu and enter the name of the message file next to "External Message File".

Group exercise
Add the question text and the French translations for the question text, labels, value sets and error messages for
marital status (B13).

Using Occurrence Labels in Question Text
We have the following occurrence labels for the housing unit types in question FO05.

e Traditional round hut
e Detached house
e Semi-detached house
e Flat/apartment

o Improvised (kiosk/container)
We can use these occurrence labels in the question text for FO5:
How many ~~getocclabel()™~~ units are in this household?

Anytime you use ~~getocclabel()~~ in question text it is replaced by the occurrence label of the current occurrence. With
the above, the question text for the first occurrence will be "How many traditional round hut units are in this household?"
and the text for the second occurrence will be "How many detached house units are in this household?" ...

Case Labels

By default, the case listing screen shows the ID items concatenated together. This is not very easy for an interviewer to
read. You can customize the case listing for a case using the setcaselabel command. As an example let's set the case
label to the string "province-district-ea-household number: name of head of household". Since we need to have the name of
the head of household to do this, we can add it in the postproc of NAME. In order to format the case label from the variables
we can use the function maketext () which works like errmsg () but returns a string instead of displaying it on the
screen.

PROC NAME

if curocc() = 1 then
// Set label for case in case listing to an easier to read format.
// We do this when we first get the name of the head.
string caselLabel = maketext("%d-%02d-%03d-%03d: %s",
PROVINCE, DISTRICT, ENUMERATION_AREA,
HOUSEHOLD_NUMBER, strip(NAME));
setcaselabel (POPSTAN2020_DICT, caselabel);
endif;

The %02d and %03d tell CSPro to format the numbers with 2 and 3 zero-filled decimal digits respectively. If DISTRICT is 1 this
will result in "001" instead of just "1". The sizes 2 and 3 are based on the lengths of these variables in the dictionary. Instead
of using %d we can use %V which will automtically use the length and zero-fill setting of the variables from the dictionary to
format the values.

PROC NAME

if curocc() = 1 then
// Set label for case in case listing to an easier to read format.
// We do this when we first get the name of the head.
string caselLabel = maketext("%v-%v-%v-%v: %s",
PROVINCE, DISTRICT, ENUMERATION_AREA,
HOUSEHOLD_NUMBER, strip(NAME));
setcaselabel (POPSTAN2020 DICT, caselabel);
endif;

Now after entering a case we have a much friendlier case listing:

"W CSEntry (Application: HouseholdQuestion

File Mode Edit Mavigation View Opti
| @ |[Ba B |1 B« | R

File
e ) 1-1-001-001: John Brown

...[_] 1-1-001-002: Sam Smith

Note that setcaselabel only works with data files of type CSPro DB.

Dynamic Value Sets

It is often useful to change the value set for a question from logic. This can be done using the command setvalueset.
Let's start with a simple example. Currently our value set for relationship in section B has labels like "Son/Daughter" and
"Brother/Sister" to allow for both genders. However, when we show the relationship value set we already know the gender
of the household member so we could show "Son" for males and "Daughter" for females. To do this we create two new
value sets for relationship in the dictionary: RELATIONSHIP_MALE and RELATIONSHIP_FEMALE. Then we can use logic to
choose between the value sets depending on the value of SEX. Which proc do we do that in? We do that in the onfocus of
RELATIONSHIP since we need to use it when we are in that field. The onfocus is called every time the field is entered. We
cannot do this in the preproc since the preproc is not triggered when moving backwards through the questions.

PROC RELATIONSHIP
onfocus
// Show male or female version of value set depending on sex of the person.
if SEX = 1 then
setvalueset (RELATIONSHIP, RELATIONSHIP_MALE);
else
setvalueset (RELATIONSHIP, RELATIONSHIP_FEMALE);
endif;

Dynamic Value Sets from a Roster

For question Al1, line number of respondent, we would like to create a value set from the names and line numbers of the
eligible household members. For this to work correctly we will need to ask A1l after entering the names and agesin the
household members rosters. Create a new numeric variable in the dictionary for A11 and place it on the household members
form after the roster.

We will fill in the ValueSet object with names and line numbers of the eligible members in the household. According the
specification, only household members 12 and over can be respondents. For example, if we have the following household:

Sex| Relationship| Ag

[i7]

Line number
[ 2]
4]

NI

We would fill in the ValueSet object as follows:

Subscript Codes Labels

1 1 John Brown
2 2 Mary Brown
3 4 Jane Brown

To create the value set from the household roster we need the second form of setvalueset that takes a ValueSet object. This
will allow us to create the list of names in logic instead of in the dictionary.

First, we need to declare a ValueSet object. A ValueSet object is also a variable like a string or numeric but instead of holding
just a single value it can hold and entire value set.

PROC RESPONDENT_LINE_NUMBER
onfocus
ValueSet eligibleMembersVSet;

We can add values to the value set by calling the function eligibleMembersVSet.add (). The "." between

"eligibleMembersVSet" and "add" tells CSPro to apply the add operation to the ValueSet before the "." which in this case is
the ValueSet named eligibleMembersVSet. For example to add John Brown to the value set we would do the following:

eligibleMembersVSet.add("John Brown", 1);
So to build our value set we need to call add for each member of the household 12 and over.

eligibleMembersVSet.add("John Brown", 1);
eligibleMembersVSet.add("Mary Brown", 2);
eligibleMembersVSet.add("Jane Brown", 4);

To make this work for any household we need to call add in loop over the household members.

PROC RESPONDENT_LINE_NUMBER

onfocus

// Create the value set for respondent from all household members 12 and over

ValueSet eligibleMembersVSet;

do numeric indexRoster = 1 while indexRoster <= totocc(HOUSEHOLD MEMBERS ROSTER)
if AGE(indexRoster) »>= 12 then

eligibleMembersVSet.add(NAME(indexRoster),indexRoster);

endif;

enddo;

setvalueset (RESPONDENT_LINE_NUMBER, eligibleMembersVSet);

Finally, make sure to set the capture type of the field to "Radio Button" in the field properties for
RESPONDENT_LINE_NUMBER.

As a second example we will implement the dynamic value set for mother line number (B10). For this value set we only show
females in the household over 12 and don't allow a woman to be her own mother.

// Create value set from women over 12 in household roster
numeric indexRoster;
ValueSet motherVSet;

do numeric indexRoster = 1 while indexRoster <= totocc(HOUSEHOLD_MEMBERS_ROSTER)
if SEX(indexRoster) = 2 and AGE(indexRoster) > 12 and indexRoster <> curocc() then
motherVSet.add(NAME (indexRoster),indexRoster);

endif;
enddo;

We also need to include the codes for non-resident and deceased in the value set.

motherVSet.add("Non-resident", 87);
motherVSet.add("Deceased", 88);

Finally we need to apply the value set to the field.

setvalueset (MOTHER_LINE_NUMBER, motherVSet);

Dynamic Value Sets from Checkboxes

Question B16 (primary language) should only show a subset of the languages chosen in B15 (languages spoken). We can do
this using a dynamic value set as well. The trick is that since B15 uses checkboxes it will have alpha codes (A, B, C...) while
B16 will have numeric codes (1,2,3...) so we need to convert from numeric to alpha to determine if a given language was
selected. We could do this with a series of if then else statements but an easier approach is to use the string
"ABCDEFGH" to convert from numeric to alpha by looking up the character at the position of the numeric code. For
example, numeric code 1 would give us the character at the first position: A. Numeric code 2 would give us the character at
position 2, B etc...

PROC MAIN_ LANGUAGE
onfocus

// Create value set from items selected in languages spoken
// Used to translate from checkbox (alpha codes) to numeric codes
string languageCheckboxCodes = "ABCDEFGH";

ValueSet languageVSet = MAIN_LANGUAGE_VS1;

// Loop through the numeric codes 1-8 and add each selected
// to value set
do numeric languageNumericCode = 1 while languageNumericCode <= 8

// Convert the numeric code to the checkbox alpha code
// by looking it up in the string.
string languageCheckboxCode = languageCheckboxCodes[languageNumericCode:1];

// Check if the language is selected in the checkbox field

if pos(languageCheckboxCode, LANGUAGES_SPOKEN) = © then
// Language is not selected. Remove it from the value set.
languageVSet. remove(languageNumericCode);

endif;

enddo;

// Modify value set
setvalueset (MAIN_LANGUAGE, languageVSet);

What if the interviewer doesn't pick any language in B15? Then our dynamic value set is empty. We should add a check to
B15 to ensure that at least one language is chosen.

PROC LANGUAGES_SPOKEN

// Ensure that at least on language is chosen

if length(strip(LANGUAGES SPOKEN)) = © then
errmsg("You must choose at least one language");
reenter;

endif;

Dynamic Checkboxes

Let's implement a dynamic value set for question G2, "were assets purchased with a loan". Rather than a series of yes/no
questions, we implement this using a single variable with checkboxes. We could have one checkbox for each of the 10 items
in the assets roster but it would be better if we only displayed the checkboxes for the assets that the household actually
possesses. How do we know if the household possesses an item? The household possesses the item if its quantity is greater
than zero. We need to loop through the rows of the roster and add a checkbox to the value set for each item with quantity
greater than zero. The only tricky part is that these are checkboxes so we need to use alpha values.

In order to create a value set with alpha values we need declare a ValueSet object that accepts alpha codes.

ValueSet string possessionsVSet;

We build the value set in the onfocus of the checkboxes field. We use the alphabet string trick again to get the alpha codes
from the occurrence number. We also use the function getocclabel () to get the occurrence label from the assets roster
to use in the value set.

PROC POSSESSIONS_PURCHASED WITH_LOAN
onfocus

// Create dynamic value set from assets that have quantity > ©
ValueSet string possessionsVSet;

string alphabet = "ABCDEFGHIJ";

do numeric assetNumber = 1 while assetNumber <= totocc(POSSESSIONS ROSTER)
// Check if household possesses this asset
if QUANTITY(assetNumber) > © then
// Add to value set
string label = getocclabel (POSSESSIONS_ROSTER(assetNumber));
possessionsVSet.add(label, alphabet[assetNumber:1]);
endif;
enddo;

setvalueset($, possessionsVSet);

Exercises

1. Add question text to questions BO6 through B8. Use fills to include the name as we did in the examples.

2. Add a new language, the language of your choice, to the CAPI text and to the dictionary. Translate the question text,
labels and value sets questions BO6-B08 into the new language.

3. Add question text for section G and use the occurrence labels to fill in the name of the possessions in the quantity and
value fields.

4. In question BO6 (date of birth) use a dynamic value set for the day based on the month (January: 1-31, February: 1-28,
March: 1-31...) so that the interviewer cannot enter an invalid date like February 30 or April 31. Bonus if you can
correctly handle leap years.

5. For EO8 (line number of mother of deceased) use a dynamic value set to list the names of all eligible women from the
household roster.

Session 06: Functions, Lookup Files, Navigation &
System Control

At the end of this session participants will be able to:

o Use lookup files in data entry

e Create user-defined functions in CSPro

e Extend the interface of CSEntry with userbar buttons

e Use the commands advance and move to navigate through the questionnaire
o Use the function visualvalue to get the value of variables that are "off path"

e Use the command showarray to display tables to the interviewer

Lookup Files

It is currently possible for the interviewer to enter a district code that is not valid for the province that was selected. In order
to verify that the district code is valid for the province we need to use the list of province and district codes in the annex of
the questionnaire. We can do this by creating a lookup file from the Excel spreadsheet. We can then use this lookup file to
check for valid districts.

First, we need to create a dictionary for our lookup file. For this task, we want to be able to query if the combination of the
province and district codes is a valid district. In annex 2 of the questionnaire we have a table with the province code, district
code and district name. We can use this as a lookup file where the keys are the province and district code and the value is
the district name. We can create a dictionary for this using the Excel to CSPro tool from the tools menu. Select the "Create
CSPro dictionary from Excel File" tab, click "Select Excel File" and browse to the file QuestionnaireAnnexes.xIsx. Choose
"ANN2 — District Codes" and click "Analyze Worksheet". This detects the dictionary variables to create from the columns of
the spreadsheet. The tool will show four variables, one for each column and will set the default names to column headers of
the spreadsheet. Our lookup file dictionary will have the province code and district code as the ID items and province and
district names as regular variables. Next to the province and district codes check the "ID" box to indicate that these should
be ID items. In order to avoid name conflicts with the main dictionary we can use DI_PROVINCE_CODE, DI_DISTRICT_CODE,
DI_PROVINCE_NAME, and DI_DISTRICT_NAME as the variable names. Modify them in the Excel to CSPro tool.

B Excel to CSPro _ o w
File  Help
Excelto C5Prg  Create CSPro Dictionary from Excel File

Excel File: Create CSPro Dictionary

Questionnaire Annexes xls .
Worksheet: | ANN 2-District Codes ~ Select Excel File Name prefix: DISTRICT | I Create Dictionary
Starting row: Analyze Worksheet Zero fil numerics Use explicit decimal character

Dictionary Contents

Include Name ID  Numeric Alpha Length Integer Length  Decimal Length  Create Value Set
oo | |

1 |T ‘ | | [] 7 Discrate Valueis)
oo ]| | |

2 |2 ‘ | | [] &0 Discrete Valus(s)

We must make sure that the lengths and zero-fill settings of the ID items exactly match those in the main dictionary
otherwise we won't be able to use the variables from the main dictionary as keys for the lookup. Verify that the length of the
province and district codes match those in the main dictionary and check the box "Zero fill numerics" to make sure that the
ID items are zero filled. Once all settings are correct, click create dictionary to generate the CSPro dictionary file. Save it as
"Districts.dcf" in the household folder.

2 3
File  Help

Excelto C5Pro  Create CSPro Dictionary from Excel File

Select Files Options
Excel file: Questionnaire Annexes xls Select Excel File Starting row:
| DISTRICT_DICT: Districts.def Select CSPro Dictionary Case management: | Create new file ~
| ) o .
| Output catafie: Districs codb Select Output Data Fis | [[] ok Processing fthe C37ro output data e o
Create C5Pro Data File
| Record to Workshest Mapping ftem to Calumn Mapping
| DISTRICT Record DISTRICT_REC
|
I DISTRICT_REC ANN 2-District Codes v FROVINCE_CODE
PROVINCE_CODE Column B: Province Code ~
| Reset Mappings Assign Default Mappings
| DISTRICT.CODE
i DISTRICT_CODE Column D: District Code ~
| PROVINCE_NAME
PROVINCE_NAME Column A: Province Mame ~
DUISTRICT_NAME
DISTRICT_NAME Column C: District Name ~

Once we have the dictionary we need to convert the Excel spreadsheet into a CSPro data file. We can use the first tab of the
Excel to CSPro tool to do this. Select the QuestionnaireAnnexes.xlsx file again along with the Districts dictionary that we just
created. Specify a new output data file to write the lookup file to. Under "Record to Worksheet Mapping" choose the second
worksheet "ANN 2-District Codes". Under the "Item to Column Mapping" link the dictionary variables to the appropriate
columns in the Excel spreadsheet: DI_PROVINCE_CODE to "Column B: Province Code", DI_DISTRICT_CODE to "Column D:
District Code", etc... Finally, click "Create CSPro Data File" to generate the file. Save it in the Data folder. Verify the file in
Data Manager to make it sure it was converted correctly.

Finally, in the district proc use the loadcase () command to lookup the province and district codes in the file. Loadcase
takes the name of the dictionary (DISTRICT_DICT) and the values to use as keys (ID items) for the lookup. For example, to
lookup province 3, district 6 we would do:

loadcase(DISTRICT_DICT, 3, 6)

If loadcase finds a record in the lookup file with province code 3 and district code 6 it will return 1 and set the variables in
DISTRICT_DICT to the values from the case it found. In this case that means setting the two ID items DI_PROVINCE_CODE,
DI_DISTRICT_CODE and the variables DI_DISTRICT_NAME and DI_PROVINCE_NAME.

We can use this to test if the province and district codes are valid in the DISTRICT proc:

PROC DISTRICT

// Verify that the district code is valid for the province selected.
if loadcase(DISTRICT _DICT, PROVINCE, DISTRICT) = © then
errmsg("District code %d is not valid for province %1",
DISTRICT, PROVINCE);
reenter;
else
errmsg("You have selected district: %s", DI_DISTRICT_NAME);
endif;

Note that we are using the PROVINCE and DISTRICT from the main dictionary as arguments to 1oadcase, not the ID items

from the districts dictionary. Before calling loadcase the ID items for the external dictionary are all blank. They are only set if
loadcase is successful.

Note that when you run this application, in addition to copying the pen and pff files to the Android device, you must now also
copy the lookup file (the .csdb file).

We can now add alpha variables to the main dictionary, assign the province and district names to them and display them on
the form as protected fields:

PROC DISTRICT

// Verify that the district code is valid for the province
// selected.
if loadcase(DISTRICT_DICT, PROVINCE, DISTRICT) = © then
errmsg("District code %d is not valid for province %1",
DISTRICT, PROVINCE);
reenter;
else
// Assign province and district names from lookup to main
// dictionary variables so we can display them on form.
PROVINCE_NAME DI_PROVINCE_NAME;
DISTRICT_NAME DI _DISTRICT_NAME;
endif;

Updating a Lookup File

If the data for your lookup file changes once your survey is in the field, rather than completely regenerating the lookup file
you can update the existing lookup data file based on a new Excel spreadsheet. Choose the existing csdb file as the output
data file and under "Case Management" choose "Modify, add cases" or "Modify, add, delete cases". Instead of erasing and
regenerating the data file this will update only the cases that are new or have been modified in the Excel file. Choosing
"Modify, add, delete cases" will also delete cases that are in the data file but are not in the spreadsheet. Using this approach
is important if you are planning to use CSPro data synchronization to update the lookup file during data collection. If you are
using data synchronization to update the file and you generate a completely new data file you will get duplicate cases when
the new lookup file is downloaded to the devices. If instead, you update the existing file then data synchronization will simply
update the data files on devices that download it.

Dynamic Value Set from a Lookup File

An alternative to using the lookup file to check the district and province codes is to combine the lookup file with
setvalueset to create a dynamic value set for the district that includes only districts in the selected province. To do this
we need to extract all the districts in the selected province from the lookup file. We can do this using the forcase loop
which iterates over all cases in an external data file. By itself, forcase will loop through each case in the data file. You can
also add an optional "where" clause to only go through the districts in the selected province.

PROC DISTRICT

onfocus

// Create dynamic value set of districts for selected province using lookup file

ValueSet districtsVSet;

forcase DISTRICT DICT where DI_PROVINCE CODE = PROVINCE do
districtsVSet.add(DI_DISTRICT_NAME, DI_DISTRICT CODE);

endfor;

setvalueset (DISTRICT, districtsVSet);

User-defined functions

Often you find you have identical blocks of logic in multiple procs in your application. This can cause problems if you later
change the code in one place to fix a bug and forget to change it in the other. In such situations it is better to put the logic in
a user-defined function which you can then call from all the procs where it is used. User-defined functions are defined in the
PROC GLOBAL. Anything declared in the PROC global is available in all the PROCs in your program. If you declare a logic
variable inside the PROC of a dictionary variable or group, it is only available inside that PROC but if you declare it in the
PROC GLOBAL you can use it anywhere. To view the proc GLOBAL, in the logic view, click on the first item in the form tree.
This shows all of the program logic at once: the proc GLOBAL plus all the other procs. Clicking on any other item in the form
tree shows just the procs for that item.

User-defined functions can take arguments and return values just like built in functions. You call them the same way you call
built in CSPro functions. Let's define a function that we can use in all the places where we check if a household member is a
woman of childbearing age. This function will be passed the index (row number) of the household member in the household
roster and it will return one if the person is a woman over 12 years old and zero otherwise. This way if we later decide that
we should be using 13 or 14 as a minimum age we only have to make the change in one place.

// Determine if member of household from household roster is a woman of childbearing
// age. Pass in the index (occurrence number) of the household member.
function isChildbearingWoman(index)
if SEX(index) = 2 and AGE(index) »>= 12 then
isChildbearingWoman = 1;
else
isChildbearingWoman = 0;
endif;
end;

Now we can use this function when building the value sets for line number of mother of child (B10) and the line number of
mother of deceased (E08).

PROC MOTHER_LINE_NUMBER

onfocus

// Create the value set for child mother from all eligible women

// in household roster.

ValueSet motherVSet;

do numeric indexRoster = 1 while indexRoster <= totocc(HOUSEHOLD_ MEMBERS_ROSTER)
if isChildbearingWoman(indexRoster) = 1 and indexRoster <> curocc() then

motherVSet.add(NAME (indexRoster), indexRoster);

endif;

enddo;

motherVSet.add("Non-resident",87);

motherVSet.add("Deceased",88);

setvalueset (MOTHER_LINE_NUMBER, motherVSet);

PROC MOTHER_OF_DECEASED_LINE_NUMBER
onfocus
// Create the value set for deceased mother from all eligible women
// in household roster
ValueSet motherVSet;
do numeric indexRoster = 1 while indexRoster <= totocc(HOUSEHOLD_ MEMBERS_ROSTER)
if isChildbearingWoman(indexRoster) = 1 then
motherVSet.add(NAME (indexRoster), indexRoster);
endif;
enddo;
motherVSet.add("not in household",99);
setvalueset (DECEASED MOTHERS LINE NUMBER, motherVSet);

Looking at these two procs we could probably move the shared code that creates the value set into a function as well.

// Create a value set of all household members that are eligible to be
// mothers by filling in the ValueSet motherVSet that is passed as
// the first argument.
// Labels are names of household members and codes are the corresponding
// line numbers.
// The household member in row excludeIndex will not be included in the value
// set. This can be used to exclude someone from being their own mother.
function createMothersValueSet(ValueSet motherVSet, excludeIndex)
do numeric indexRoster = 1 while indexRoster <= totocc(HOUSEHOLD MEMBERS ROSTER)
if isEligibleMother(indexRoster) = 1 and indexRoster <> excludeIndex then
motherVSet.add(NAME (indexRoster), indexRoster);
endif;
enddo;
end;

PROC MOTHER_LINE_NUMBER
onfocus

// Create the value set for child mother from all eligible
// women in household roster. Exclude current occurrence so
// that person cannot be their own mother.

ValueSet motherVSet;

createMothersValueSet(motherVSet, curocc());

// Add additional entries for non-resident and deceased
motherVSet.add("Non-resident",87);
motherVSet.add("Deceased",88);

// Update the value set with values in ValueSet object
setvalueset (MOTHER_LINE_NUMBER, motherVSet);

PROC MOTHER_OF_DECEASED_LINE_NUMBER

onfocus

// Create the value set for child mother from all eligible
// women in household roster

ValueSet motherVSet;

createMothersValueSet(motherVSet, 0);

// Add additional entry for not in household
motherVSet.add("not in household",99);

// Update the value set with values in codes and labels
setvalueset($, mothervSet);

Functions in Question Text

Another use for user-defined functions is for fills in question that require some calculation. For example, for question D05 we
want to double check that the total of the children living with the women, living elsewhere and deceased equals the number
of total births. We do this by asking the question:

Just to make sure that | have this right, (name) has had in total (total number) births during her life. Is this correct?

We can use the question text for this but we don't have a dictionary variable for total births. We only have the yes/no
variable IS_TOTAL_BIRTHS_CORRECT. We could create an additional dictionary variable but instead we can simply create a
function to compute the total births and use that as the fill value.

// Compute totalBirths to display in question text in fertility section
function totalBirths()

totalBirths = CHILDREN_IN_HOUSEHOLD + CHILDREN_ELSEWHERE + CHILDREN_DECEASED;
end;

In the question text we can simply add the function call surrounded by ~~:

Just to make sure that | have this right, ~~NAME~~ has had in total ~~totalBirths()~~ births during her life. Is this
correct?

What happens when one of the fields in this calculation is skipped? The value becomes notappl which messes up the entire
calculation. We need to be a little smarter in calculating our total to exclude the skipped values.

// Compute totalBirths to display in question text in fertility section
function totalBirths()

numeric total = 9;

if CHILDREN_IN HOUSEHOLD <> notappl then
total = total + CHILDREN_IN HOUSEHOLD;
endif;

if CHILDREN_ELSEWHERE <> notappl then
total = total + CHILDREN_ELSEWHERE;
endif;

if CHILDREN_DECEASED <> notappl then
total = total + CHILDREN_DECEASED;
endif;
totalBirths = total;
end;

The Userbar

From logic we can add buttons to the CSEntry user interface that call user-defined functions. Here is how to add a userbar
button that will create an errmsg dialog that says "hello". First we define the function hello in the PROC GLOBAL:

function hello()
errmsg("Hello");
end;

Then in the preproc of the application we add it to the userbar:

PROC POPSTAN2020_ FF

preproc

userbar(clear);

userbar(add button, "Hello", hello);
userbar(show);

When adding a button in the preproc of the application it is important to call clear first, otherwise we can end up with two
or three copies of the same button if we start the application multiple times. We added the button in the preproc of the
application but you can add or remove buttons in any proc so you can, for example, only a show button within a certain field
or a certain roster.

Let's try a more interesting example. Let'sadd a "Go To..." button that will let the user navigate directly to a particular
section of the questionnaire.

// Userbar function for navigating directly to different parts of questionnaire.
function goto()

numeric section = accept("Go to?",
"Identification",
"Household members",
"Demographics");
if section = 1 then
skip to IDENTIFICATION_FORM;
elseif section = 2 then
skip to HOUSEHOLD MEMBERS_FORM;
elseif section = 3 then
skip to DEMOGRAPHICS FORM;
endif;
end;

PROC POPSTAN2020 FF

preproc

userbar(clear);

userbar(add button, "Go To...", goto);
userbar(show);

Advance and Move

The above will let the interviewer jump from one section to another but using skip means that if we use our Go To... button
to jump over a section, that section will end up skipped and won't be saved in the data file. Instead of using skip we can use
advance which moves forward in the questionnaire without marking fields as skipped. Using advance also runs all the

preprocs and postprocs of the fields that are passed through to ensure that no consistency or out of range checks are missed.

// Userbar function for navigating directly to different parts of questionnaire.
function goto()
numeric section = accept("Go to?",
"Identification",
"Households",
"Demographics");
if section = 1 then
advance to IDENTIFICATION_FORM;
elseif section = 2 then
advance to HOUSEHOLD_MEMBERS_FORM;
elseif section = 3 then
advance to DEMOGRAPHICS_FORM;
endif;
end;

This stops the function from skipping over data when navigating, however it still has a limitation. We can only navigate
forward in the questionnaire. To go backwards we need to use reenter, but in our goto() function we don't know if the
user wants to move forward or backward. Fortunately, CSPro provides the command move which will use either skip or

reenter as appropriate. By default, when going forward, move does a skip but you can add advance after the field name
to make it do an advance instead of a skip.

// Userbar function for navigating directly to different parts of questionnaire.
function goto()
numeric section = accept("Go to?",
"Identification",
"Household members",
"Demographics");
if section = 1 then
move to IDENTIFICATION FORM advance;
elseif section = 2 then
move to HOUSEHOLD_MEMBERS_FORM advance;
elseif section = 3 then
move to DEMOGRAPHICS FORM advance;
endif;
end;

Let's extend our goto function to let the interviewer navigate directly to an individual in the household roster. If they choose
"household members" then instead of going to the first person in the roster, we will show them a list of all household
members in the roster and the let them choose which one to goto. For this we can use the function showarray (). This
function takes an array of values and displays them in a grid in a dialog box and returns the row number that the user picks.

An array logic variable is similar to a dictionary item with occurrences. A numeric array of length seven stores seven
numbers, each of which is accessed through subscripts.

Array numeric codes(7);

showarray () takes a two-dimensional array. You can think of a two-dimensional array as a grid of variables or as a matrix.
You declare a two-dimensional array the same way you declare a one-dimensional array except that you specify the size in
both dimensions: number of rows and number of columns.

Array string householdMembersArray(30, 3);

In our case we want up to 30 rows, one for each person, and we will use 3 columns so that we can display the name, sex and
relationship for each person.

When assigning a value to a two-dimensional array you specify both the row and column you want to put the value in:

// Set value in row 4, column 2
householdMembersArray(4, 2) = "This is the 4th row, 2nd column";

In our examples we will loop through the members in the household roster and add the name, sex and relationship for each
one to our array. Since this will be more than a few lines of code let's put this into a function by itself and then call it from our
goto() function.

Page 49 of 87 Session 06: Functions, Lookup Files, Navigation & System Control


https://www.csprousers.org/help/CSPro/function_statement.html
https://www.csprousers.org/help/CSPro/numeric_statement.html
https://www.csprousers.org/help/CSPro/accept_function.html
https://www.csprousers.org/help/CSPro/if_statement.html
https://www.csprousers.org/help/CSPro/if_statement.html
https://www.csprousers.org/help/CSPro/move_statement.html
https://www.csprousers.org/help/CSPro/advance_statement.html
https://www.csprousers.org/help/CSPro/if_statement.html
https://www.csprousers.org/help/CSPro/if_statement.html
https://www.csprousers.org/help/CSPro/move_statement.html
https://www.csprousers.org/help/CSPro/advance_statement.html
https://www.csprousers.org/help/CSPro/if_statement.html
https://www.csprousers.org/help/CSPro/if_statement.html
https://www.csprousers.org/help/CSPro/move_statement.html
https://www.csprousers.org/help/CSPro/advance_statement.html
https://www.csprousers.org/help/CSPro/if_statement.html
https://www.csprousers.org/help/CSPro/function_statement.html
https://www.csprousers.org/help/CSPro/showarray_function.html
https://www.csprousers.org/help/CSPro/Array_statement.html
https://www.csprousers.org/help/CSPro/showarray_function.html
https://www.csprousers.org/help/CSPro/Array_statement.html

// Show list of entries in household roster in a dialog and let interviewer pick
// one. Returns the row number of the person that was picked or zero if the
// dialog was canceled.
function pickFromHouseholdRoster()
numeric i;
do i = 1 while i <= totocc(HOUSEHOLD MEMBERS_ROSTER)
householdMembersArray (i, 1) strip(NAME(i));
householdMembersArray (i, 2) getlabel (SEX, SEX(i));
householdMembersArray(i, 3) getlabel (RELATIONSHIP, RELATIONSHIP(i));
enddo;
householdMembersArray(i, 1) = ""; // Mark end
numeric picked = showarray(householdMembersArray, title("Name", "Sex",
"Relationship"));

pickFromHouseholdRoster = picked;
end;

// Userbar function for navigating
// directly to different parts of questionnaire.
function goto()
numeric section = accept("Go to?",
"Identification”,
"Household members",
"Demographics");
if section = 1 then
move to IDENTIFICATION_FORM advance;
elseif section = 2 then
numeric index = pickFromHouseholdRoster();
if index > © then
move to NAME(index) advance;
endif;
elseif section = 3 then
move to DEMOGRAPHICS FORM advance;
endif;
end;

Note that for sex and relationship we want the value set labels so we use the function getlabel(). This function returns the
label from value set as an alpha value. It takes the name of the variable (or value set) and the value to use. For example
getlabel (SEX, 1) will return"Male". Inside our loop, getlabel (SEX, SEX(1i)) will return the label for the sex of
the ith household member.

Path and Visualvalue

Our goto() function works when we are in section B or C but when we are in section A the sex and relationship are blank.
What is going on? In system controlled mode, CSEntry keeps track of which variables are on and off the "path". Variables
that you have entered are considered "on path" but those that have been skipped, even if there was a value in them before
they were skipped, are considered "off path". As we have seen before, variables that are "off path" are considered blank
(notappl) in logic. It turns out that variables that are ahead of the current field are also considered "off path" until you
pass through them. The idea is that these fields have not yet been validated by running their preproc and postproc with the
current values of all preceding fields and therefore cannot be considered final. The effect of this is that the values of all fields
ahead of the current field in the questionnaire are notappl in logic.

Interestingly, as we can see from our current code, this only applies to numeric items. Our code works just fine for the NAME
field.

You can see which fields are "on path" by looking at the background color of the field:

e Green: on path

e Dark Gray: skipped

e White: not yet been filled in
e Light Gray: protected

Note that the field coloring scheme is different in operator controlled mode.

Fortunately, we can get the value of fields that are "off path" using the function visualvalue(). It returns whatever
value is currently visible in the field whether or not it has been skipped or is ahead of the current field. Using this for
relationship and sex in our function we get:

function pickFromHouseholdRoster()
numeric i;
do i = 1 while i <= totocc(HOUSEHOLD MEMBERS_ROSTER);
householdMembersArray(i, 1) = strip(NAME(i));
householdMembersArray(i, 2) getlabel (SEX, visualvalue(SEX(i)));
householdMembersArray(i, 3) = getlabel (RELATIONSHIP,
visualvalue (RELATIONSHIP(i)));

enddo;
householdMembersArray(i, 1) = ""; // Mark end
numeric picked = showarray(householdMembersArray,
title("Name", "Sex", "Relationship"));
pickFromHouseholdRoster = picked;
end;

All that is left now is to use the appropriate relationship for the sex of the household member:

if visualvalue(SEX(i)) = 1 then
householdMembersArray(i, 3)

getlabel (RELATIONSHIP_ MALE,
visualvalue (RELATIONSHIP(i)));
else
householdMembersArray (i, 3)

getlabel (RELATIONSHIP_ FEMALE,
visualvalue(RELATIONSHIP(i)));
endif;

Setting field values only once

Let's prefill the interview start time. We can make the field protected and set the value in the preproc just like we did with
the PERSON_NUMBER field. We can use the function systime () which returns the current time as a number formatted
according to the format specification passed in.

PROC INTERVIEW_START_HOURS

preproc

// Prefill interview start time with current time.
INTERVIEW_START_TIME = systime("HHMM");

This works the first time we visit the field but what happens when we come back to the question a minute or two later? We
only want to record this value the first time the interviewer enters the field and once it is set we don't want it to change. We
can do this by comparing the value of INTERVIEW_START_TIME to blank (notapp1l) in the preproc and only setting the
value to systime if it is blank.

PROC INTERVIEW_START_HOURS

preproc

// Prefill interview start time with current time.

if INTERVIEW_START_HOURS = notappl then
INTERVIEW_START_TIME = systime("HHMM");

endif;

But this doesn't seem to work. Why? Is INTERVIEW_START_HOURS on path when we are in the preproc of
INTERVIEW_START_HOURS? It is not. We have to get to the postproc for the variable to be on path. However, we can use
visualvalue() to get the value of the field in the preproc:

PROC INTERVIEW_START_HOURS

preproc

// Prefill interview start time with current time.

if visualvalue(INTERVIEW_START_HOURS) = notappl then
INTERVIEW START TIME = systime("HHMM");

endif;

We can also fill in the interview end time automatically using systime (). Unlike with the start time, we need to do this at
the end of the interview, i.e. in the postproc of the questionnaire.

PROC POPSTAN2020_QUEST
postproc

// Set interview end time first time end of questionnaire is reached
if INTERVIEW_END TIME = notappl then

INTERVIEW_END TIME = systime("HHMM");
endif;

We should make this field protected. However, if we do that we get an error when we don't fill it in while in section A of the
questionnaire. While we don't normally want to allow a field to be left blank, in this case we need to make an exception. We
can do that by changing the validation method in the field properties to "Allow out of range without confirmation". With this
setting, you are allowed to leave a field blank or to enter a value outside the value set with no error.

Having captured the interview start and end time as hours and minutes we can calculate the total interview time by
subtracting the start time from the end time. However, this calculation is tricky because we have to handle the case where
the end minutes are less than the start minutes. It would be easier if instead of using systime () we used the function
timestamp () which gives the time in total seconds since January 1, 1970. If we subtract the start timestamp from the end
timestamp we get the total interview time in seconds. Let's add new protected variables for start timestamp, end
timestamp and interview duration and fill them in. Since these fields will be on the section A form but will not be filled until
the end of the questionnaire, set the validation method to "Allow out of range without confirmation" as we did with
INTERVIEW_END_TIME.

PROC INTERVIEW_START_TIMESTAMP

preproc

// Prefill interview start time with current time.

if visualvalue(INTERVIEW_START_TIMESTAMP) = notappl then
INTERVIEW _START TIMESTAMP = timestamp();

endif;

PROC POPSTAN2020 QUEST
postproc

// Set interview end time and total time first time end of questionnaire is reached
if INTERVIEW_END TIMESTAMP = notappl then
INTERVIEW_END TIMESTAMP = timestamp();

INTERVIEW DURATION MINUTES = (INTERVIEW END TIMESTAMP - INTERVIEW START_TIMESTAMP)/6
H
endif;

If we don't need to display the start and end times on the form we can avoid using visualvalue. We can simply remove the
variables from the form and keep them in the dictionary. When we set the values of these variables in logic and the case is
saved the values are saved to the data file just like variables that are on the form. Unlike variables on the form, however,
variables that are not on the form are always considered "on path" so you do not need to worry about using visualvalue.

Page 52 of 87 Session 06: Functions, Lookup Files, Navigation & System Control


https://www.csprousers.org/help/CSPro/proc_statement.html
https://www.csprousers.org/help/CSPro/preproc_statement.html
https://www.csprousers.org/help/CSPro/if_statement.html
https://www.csprousers.org/help/CSPro/visualvalue_function.html
https://www.csprousers.org/help/CSPro/special_values.html
https://www.csprousers.org/help/CSPro/if_statement.html
https://www.csprousers.org/help/CSPro/systime_function.html
https://www.csprousers.org/help/CSPro/if_statement.html
https://www.csprousers.org/help/CSPro/systime_function.html
https://www.csprousers.org/help/CSPro/proc_statement.html
https://www.csprousers.org/help/CSPro/postproc_statement.html
https://www.csprousers.org/help/CSPro/if_statement.html
https://www.csprousers.org/help/CSPro/special_values.html
https://www.csprousers.org/help/CSPro/if_statement.html
https://www.csprousers.org/help/CSPro/systime_function.html
https://www.csprousers.org/help/CSPro/if_statement.html
https://www.csprousers.org/help/CSPro/systime_function.html
https://www.csprousers.org/help/CSPro/timestamp_function.html
https://www.csprousers.org/help/CSPro/proc_statement.html
https://www.csprousers.org/help/CSPro/preproc_statement.html
https://www.csprousers.org/help/CSPro/if_statement.html
https://www.csprousers.org/help/CSPro/visualvalue_function.html
https://www.csprousers.org/help/CSPro/special_values.html
https://www.csprousers.org/help/CSPro/if_statement.html
https://www.csprousers.org/help/CSPro/timestamp_function.html
https://www.csprousers.org/help/CSPro/if_statement.html
https://www.csprousers.org/help/CSPro/proc_statement.html
https://www.csprousers.org/help/CSPro/postproc_statement.html
https://www.csprousers.org/help/CSPro/if_statement.html
https://www.csprousers.org/help/CSPro/special_values.html
https://www.csprousers.org/help/CSPro/if_statement.html
https://www.csprousers.org/help/CSPro/timestamp_function.html
https://www.csprousers.org/help/CSPro/if_statement.html

Exercises

1. Modify the question text for EO1 to say "Has any member of this household passed away in the past years, that is
since (year)?" where year is replaced by the year of the interview minus 5. For example, if the year of the interview is
2018 it should read "Has any member of this household passed away in the past years, that is since 2013?". Use the
the interview date from question A6 to compute the value to insert.

2. Extend the goto() function to include the remaining sections of the questionnaire (D through G).

3. Add a button to the userbar called "household summary" that when clicked uses the errmsg function to display a
message that shows the name of the head of household, and the number of household members by sex. For example:
"Head of Household: John Brown, Total Members: 5, Women: 2, Men: 3". Bonus if you can get this to work when you
click the button from section A. Hint: count and seek will not work with visualvalue so you will need to use a loop to
find the number of males and females.

4. Implement a check on the minimum and maximum per unit possession values in question GO1. Use the spreadsheet in
annex 4 to create a lookup file containing the asset code, minimum value and maximum value. Use the loadcase
command with this lookup file to find the minimum and maximum values for the selected asset and show an error
message if the per unit asset value entered in the roster is below the minimum value or above the maximum value.
This should be a soft check.

5. Fillin the interview start and end date automatically using the sysdate() command and make the interview start and
end date fields protected.

6. Add question A10, interview status to the dictionary and form. We want to fill in this question at the start of the
interview if the response is code 2 (non-contact), 3 (vacant) or 4 (refused) and then immediately end the
questionnaire without going into any of the subsequent questions. However, if there is a respondent willing to give
the interview, then the interview status should be set 5 (partially complete) until the entire questionnaire is complete
at which point it should be set to 1 (complete). To implement this, keep the question in its current position in the
guestionnaire but use a dynamic value set to limit the options so that the interview cannot be set as completed until
the entire questionnaire has been completed. The first time the field is entered (before any value has been entered)
the value set should be: 2 Non-contact, 3 Vacant, 4 Refused, 5 Continue interview. If the interviewer chooses 2, 3 or
4, end the interview, otherwise, if they choose 5, continue to the next field. At the end of the interview (postproc of
the level), if the value is currently 5, set it to 1 (complete). If the interviewer returns to A9 after the field has been set
to 1 (complete) then display the value set asit is on the questionnaire.

Session 07: Multimedia & GPS

At the end of this session participants will be able to:

e Useimagesinvalue sets

o View external documents and media files from a data entry application
e Take photosin a data entry application

e Capture GPS coordinates in a data entry application

o Display captured GPS points on a map

e Generate and view reports from a data entry application

Value Set Images

In addition to showing text in a value set we can also display images. To add an image to a value set, select the variable in
the dictionary editor and click on the "..." button next to the value. Browse to pick an image file. CSPro supports standard
image file formats: png, jpeg, bmp... We recommend using jpeg as it tends to have smaller file sizes. The images are not
added to the pen file by default so you would need to copy them to the device along with the pen file.

Let's add images for the roof types in FO8. Copy the Rooflmages folder into the application directory. Now open the value set
for FO8 and add each image to its respective value. We do not have a picture for other (specify) so we will leave that one
without an image. Run the application on Windows and on mobile and see the images. Note that on mobile we have to copy
the images to the device since they are not part of the pen file by default.

The Resource Folder

Instead of copying the image files to the mobile device each time they are updated we can put them in the resource
directory of the application. All files in the resource folder are built into the pen file and extracted on the device when the
application is run. This makes it more convenient to distribute an application with additional files. Let's create a resource
folder, add it to our application and put our image files in it. To create the resource file just create a new folder in Windows
explorer. We will create the folder "resources" in the household folder. Then in CSPro go to Add Files... from the File menu,
choose "resource folder" and browse to the new folder. Now put the whole Rooflmages folder in the resource folder and
rebuild the pen file.

Viewing External Files

The function view() can be used to launch files with their default viewer. The view() function takes the path to the file to
display. The device must have an application installed capable of viewing the type of file specified. The function determines
which application to launch to view the file based on the file extension. Most devices have built-in viewers for most photo,
music, and video file formats, however not all have a built-in PDF viewer. If you do not have one on your device, you can
always download one. The view( ) function can also display websites if you pass it a URL.

view("/mnt/sdcard/csentry/picture.jpg");
view("/mnt/sdcard/csentry/audio.mp3");
view("/mnt/sdcard/csentry/movie.mp4");
view("/mnt/sdcard/csentry/document.pdf");
view("https://www.census.gov/data/software/cspro.html");

Let's add a userbar button to show the interview manual as an aid to the interviewer :

// Display the interviewer manual

function showInterviewerManual()
view("/mnt/sdcard/csentry/Popstan2020/Housing/resources/InterviewerManual.pdf");

end;

We can make our code a little more flexible by using the function pathname () to get the directory where the application is
stored. This way if someone copies it into a different folder on the phone/tablet it will still work:

// Display the interviewer manual
function showInterviewerManual()

view(pathname(Application) + "resources/InterviewerManual.pdf");
end;

Taking Photos on Android
To take a picture on Android you can use the function execsystem which is another way to launch external applications.
To use execsystemto take a photo passit a string with "camera:" followed by the full path in which to save the photo:

execsystem("camera:/mnt/sdcard/csentry/photo.jpg");
Taking photos is currently only supported on Android. execsystem on Windows does not support the camera.

Note that photos do not get saved into the application data file. You need to save them to a folder/file on the device. The
filename you use should make it easy to link the photo back to the interview that it was taken for. The easiest way to do this
is to make the name of the photo based on the ID items of the questionnaire:

string photoFilename = pathname(application) +
maketext("../Data/HouseholdPhotos/photo%v%sviviv.jpg",
PROVINCE, DISTRICT, ENUMERATION_AREA,
HOUSEHOLD_NUMBER) ;
execsystem(maketext("camera:%s", photoFilename));

Note that we are putting the photo in a subdirectory of the data directory named HouseholdPhotos. We will need to create
this folder on the tablet for this to work.

Let's have the interviewer take a photo of the household being interviewed in section A. Rather than adding a userbar
button which would be easy for the interviewer to forget, we will make it part of the questionnaire by adding a dummy
variable to the dictionary. In the postproc of the variable we will call execsystem() to take the photo. We can add a
refused code in case the household doesn't want a picture taken.

PROC PHOTO

if PHOTO = 1 then
// Take photo
string photoFilename = pathname(application) +
maketext("../Data/HouseholdPhotos/photo%visviviv.jpg”,
PROVINCE, DISTRICT, ENUMERATION_AREA,
HOUSEHOLD_NUMBER) ;
execsystem(maketext("camera:%s", photoFilename));
elseif PHOTO = 9 then
// refused, move to next field with no photo
endif;

This works the first time through the questionnaire but if we go back through the photo question a second time then we are
forced to take another photo. We can avoid this by adding another option to the value set "Keep photo" that the
interviewer will select once they are happy with the photo that they took. When "keep photo" is selected we won't launch
the camera. We can add another option "View photo" to let them look at the photo that was taken.

PROC PHOTO

string photoFilename = pathname(application) +
maketext("../Data/HouseholdPhotos /photo%viviviv.jpg",
PROVINCE, DISTRICT, ENUMERATION_AREA,
HOUSEHOLD NUMBER) ;

if PHOTO = 1 then
// Take/retake photo
execsystem(maketext("camera:%s", photoFilename));

// reenter so that interview can retake if it is not good.
reenter;
elseif PHOTO = 2 then
// view existing photo
view(photoFilename);

// reenter so that interview can retake if it is not good.
reenter;
elseif PHOTO = 3 then
// Keep photo - move to next field
elseif PHOTO = 9 then
// No photo (refused)
// Delete photo if it exists
filedelete(photoFilename);
endif;

This works except that it is possible for the interviewer to select "Keep photo" even when there is no existing photo and to
attempt to view a photo that doesn't exist. We can solve this by only showing the "Keep photo" and "View photo" options in
the value set if the photo exists already.

PROC PHOTO
onfocus
string photoFilename = pathname(application) +
maketext("../Data/HouseholdPhotos/photo%visviviv. jpg"”,
PROVINCE, DISTRICT, ENUMERATION_AREA,
HOUSEHOLD_NUMBER) ;

// Only show the view and keep options in value set
// if the photo exists.
if fileexist(photoFilename) then
setvalueset(PHOTO, PHOTO VS PHOTO TAKEN);
else
setvalueset (PHOTO, PHOTO_VS_NO_PHOTO);
endif;

Capturing GPS Points
To capture a GPS point first you need to start up the GPS hardware:

gps(open);
Then you request a GPS reading giving it a timeout in seconds and optionally a desired accuracy in meters:

if gps(read, 60, 10) then // Read up to 60 seconds, try for 1@m accuracy
errmsg("Latitude is %f, longitude is %f",gps(latitude),gps(longitude));
else
errmsg("GPS signal could not be acquired");
endif;

Finally, you close the GPS:

gps(close);

In addition to querying the GPS latitude and longitude you can also query the accuracy, number of satellites, and altitude
(although on Android altitude is not very accurate). See the help on GPS for details.

The longitude and latitude reported are in degrees in the WGS84 datum and are returned as decimal numbers.

To save the GPS coordinates in our data file we need to add the appropriate variables to our dictionary. To store the
coordinates so that they can cover the entire earth with sufficient precision they should have space for 3 digits before the
decimal point, 1 digit for the decimal point, and at least 5 digits after the decimal and a minus sign for a minimum total
length of 10. Add the variables LATITUDE and LONGITUDE to the dictionary and to section on the forms.

Let's add a user bar button to update the GPS coordinates for the household:

// Capture current household location using GPS
function getGPS()
gps(open);
if gps(read, 60, 10) then // Read up to 60 seconds, try for 10m accuracy
LATITUDE = gps(latitude);
LONGITUDE = gps(longitude);
else
errmsg("GPS signal could not be acquired");
endif;
gps(close);
end;

Displaying GPS Points on a Map

If we want to be fancy, we can display the point on the map first so that the interviewer can view it and then decide
whether or not to use it. We can do this using a variable of type Map. Like ValueSet variables, Map variables have
functions that can be called on them using a dot ("."). For example, to add a marker to a map you would call:

Map mymap;
mymap.addMarker (longitude, latitude);

To display the map, call the show () function. This displays a Google Map containing the marker that was added.
We can use a map variable to display a map that has a marker at the GPS location that we captured:

// Capture current household location using GPS
function getGPS()
gps(open);
if gps(read, 60, 10) then // Read up to 60 seconds, try for 1@m accuracy
// Show map so that interviewer can see result
Map m;
m.addMarker (gps(latitude), gps(longitude));
m.show();
if accept("Save this result"”, "Yes", "No") = 1 then
LATITUDE = gps(latitude);
LONGITUDE = gps(longitude);
endif;
else
errmsg("GPS signal could not be acquired");
endif;
gps(close);
end;

Variables of type Map have many other functions that let you customize the map, the markers, add buttons and run logic
functions when markers are tapped. You can also use offline base maps for situations where you do not have an Internet
connection. See the help for maps for details.

Showing the Case Listing on a Map

If you have latitude and longitude variables in your data dictionary for each household you can configure the case listing
screen to display the households on a map on Android. To do this, open the Mapping Options dialog from the Options menu
in CSPro designer. Check the box "Show the case listing on a map" and choose the latitude and longitude variables from your
dictionary. When you run the application on Android, the case listing screen will show a map with a marker for each
household. Tapping on a marker launches data entry for that household.

Group Exercise
Build a CSPro application for the household listing questionnaire. Place your application in the folder
Popstan2020/Listing. Treat each household as a separate case. In other words, do not use a repeating record and
roster to try to match the paper questionnaire. When creating the listing dictionary use "LI_" as a prefix for the all
the variable names so that they will not conflict with the names in the household dictionary. There is no need to
validate the ID items using a lookup file like we did for the household questionnaire. We will prefill the ID items from
the menu program tomorrow. For each household, automatically record the GPS coordinates in the latitude field.
Give the interviewer the option of taking a photo of the household but allow for refusal. Name the photo based on
the case ID items and put the photos in the directory Popstan2020/Data/ListingPhotos. Use the Mapping Options
dialog to show the case listing on a map for the listing program. Go outside and test your listing application on a
tablet.

Writing and Viewing Reports
With the view() function you can display text files on your device. We can use this to show text documents that we write
out from CSPro. To write out files from a CSEntry application we have to first declare a variable of type file :

File tempFile;

Then we can use the commands setfile(), filewrite() and close() to open, write to and close the file. Let's
create a userbar button to write a simple text report that lists the household members:

// Write out and display household summary report
function showHouseholdReport()
string reportFilename = maketext("%sreport.txt", pathname(Application));
File tempFile;
setfile(tempFile, reportFilename, create);
filewrite(tempFile, "Household Summary Report");
filewrite(tempFile, "------------------------ "Y;
filewrite(tempFile, "");
filewrite(tempFile, "Province %d District %d EA %d Area type %d Household Number
%d",
visualvalue (PROVINCE),
visualvalue(DISTRICT),
visualvalue (ENUMERATION_AREA),
visualvalue(AREA TYPE),
visualvalue(HOUSEHOLD NUMBER));
filewrite(tempFile, "");
filewrite(tempFile, "Household Members:");
filewrite(tempFile, "");
filewrite(tempFile, "Name Sex Age Relationship");
filewrite(tempFile, "---- --- e "Y;
do numeric i = 1 while i <= totocc(HOUSEHOLD MEMBERS_ ROSTER)
filewrite(tempFile, "%s %-6s %3d %s",
NAME (i),
getlabel (SEX, visualvalue(SEX(i))),
visualvalue(AGE(i)),
getlabel (RELATIONSHIP_VS1,
visualvalue(RELATIONSHIP(i))));
enddo;
close(tempFile);

view(reportFilename);
end;

If you are familiar with HTML you can write out HTML and have nicer formatting and pictures. CSPro also has a built in
templating engine to generate HTML reports using templates. See Templated Reports in the CSPro helps for details.

Exercises

1. Add a button to the userbar that shows your website using the view() function.

2. Add the following additional information to the summary report. a. Total number of household members, total
males, total females. b. List of household assets from section G with quantity. Do not list assets where the quantity is
zero.

Session 08: Menu Programs

At the end of this session participants will be able to:

e Understand how to design the screens of a menu program
e Use variables of type pff to launch one CSPro application from another
o (Create a simple menu program to launch the main data entry program

e Create dynamic menus with dynamic value sets

What is a Menu Program?

A menu program is a CSPro data entry application that is used to manage the data entry workflow. A menu program is not
used for capturing any interview data itself. Instead, it launches other data entry programs for interviews. Menu programs
generally have some or all of following functions:

e Launches other CSPro applications to do data collection (often pre-filling ID items)

e Show reports on progress, summary statistics

e Manage user access through usernames/passwords

e Manage household listing and interview assignments

e Launch synchronization

Designing the Screens and Flow
Once you decide on which functions you want your menu program to perform, the next step is to design the screens and how
they link together. This is most easily done as a diagram like the one below for a fairly simple menu program.

Supervisor Menu

® Supervisor e Completionreport
e Interviewer e |logout

Interviewer Menu

e List households
e Interview households Completion Report

e logout

Listing Data Entry
Program

Household Data
Entry Program

Creating the Dictionary and Forms
Once you have the screens, the next step is to create a new data entry application for your menu program. Let's name ours
Menu and put it in the folder Popstan2020/Menu.

Each menu screen will be a different variable in the dictionary. The value set for the variable will show the available menu
choices to the enumerator on that screen. In the menu example above we will have three variables:

e LOGIN (value set: Interviewer - 1, Supervisor - 2)
e INTERVIEWER_MAIN_MENU (value set: List Households - 1, Interview Household - 2, Logout - 9)

e SUPERVISOR_MAIN_MENU (value set: Completion report - 1, Logout - 9).

We will keep the default ID item that CSPro creates for us. The menu program dictionary doesn't really need an ID item
since we do not need to save our menu choices to the data file. However, CSPro requires that we have at least one ID item
so we will keep it.

Create the items in the dictionary and then create a form and drop the items onto the form. Do not drop the ID item onto
the form. We can leave it off the form since we will never actually write out a completed case from the menu program.
Unlike typical data entry applications, menu programs tend not to have a linear flow. As a result, the order of the variables
on the form is less important. We will use skips and reenters to move from one menu to another.

Menu Program Logic
The logic for processing the menu choice for each screen goes in the postproc of the variable for the menu. For example, to
process the login menu field in our example we would have the following logic:

PROC LOGIN

// Go to the appropriate menu for the role chosen
if $§ = 1 then
skip to INTERVIEWER_MAIN_MENU;
else
skip to SUPERVISOR_MAIN MENU;
endif;

If the user selects to login as an interviewer we skip to the interviewer main menu field to show the interviewer menu,
otherwise we skip to the supervisor main menu field to show the supervisor menu.

Handling the interviewer menu is similar. Here we will create user-defined functions to launch the listing and household
programs.

PROC INTERVIEWER_MAIN_MENU

postproc
// Handle the menu choice
if $§ = 1 then
// List households
launchHouseholdListing();
elseif $ = 2 then
// Household questionnaire
launchHouseholdDataEntry();
elseif $ = 9 then
// Logout
stop(1);
endif;

// Show interviewer menu again
reenter;

Page 61 of 87 Session 08: Menu Programs


https://www.csprousers.org/help/CSPro/proc_statement.html
https://www.csprousers.org/help/CSPro/if_statement.html
https://www.csprousers.org/help/CSPro/if_statement.html
https://www.csprousers.org/help/CSPro/skip_statement.html
https://www.csprousers.org/help/CSPro/if_statement.html
https://www.csprousers.org/help/CSPro/skip_statement.html
https://www.csprousers.org/help/CSPro/if_statement.html
https://www.csprousers.org/help/CSPro/proc_statement.html
https://www.csprousers.org/help/CSPro/postproc_statement.html
https://www.csprousers.org/help/CSPro/if_statement.html
https://www.csprousers.org/help/CSPro/if_statement.html
https://www.csprousers.org/help/CSPro/if_statement.html
https://www.csprousers.org/help/CSPro/if_statement.html
https://www.csprousers.org/help/CSPro/if_statement.html
https://www.csprousers.org/help/CSPro/if_statement.html
https://www.csprousers.org/help/CSPro/stop_function.html
https://www.csprousers.org/help/CSPro/if_statement.html
https://www.csprousers.org/help/CSPro/reenter_statement.html

It is important to make sure that after the postproc of the menu field we do not let CSEntry continue to the next field,
otherwise after the interviewer launches the household listing they would end up in the next field, which, in this case is the
supervisor menu. To prevent this, we put a reenter at the end of the postproc so that we go back into the same menu field
again.

It looks kind of strange that when we go back into a menu, the previous choice is still selected. We can prevent this by
clearing the choice in the onfocus of the field.

onfocus
// Clear previous choice
$ = notappl;

The supervisor menu is similar to the interviewer menu:

PROC SUPERVISOR_MAIN_MENU

onfocus
// Clear previous choice
$ = notappl;

postproc
// Handle the menu choice
if $§ = 1 then
// Show report
showCompletionReport();
elseif $ = 9 then
// Logout
stop(1);
endif;

// Show supervisor menu again
reenter;

Launching one CSPro Application from Another

Let's fill in the function to launch the household data entry program. We can launch other CSPro programs from within our
data entry program using a variable of type Pff. A Pff variable represents a pff file. You can load the contents of an
existing pff file into a PFf variable using the load() function and passing it the path to the file. You can then run the
associated CSPro application by calling the exec() function. This will start the application using the parameters in the pff file.

The following logic will launch a data entry application using the pff file Popstan2020.pff in the sibling directory to the menu
program directory

function launchHouseholdDataEntry()

Pff householdPff;
householdPff.load("../Household/Popstan2020.pff");
householdPff.exec();

end;

This will start the data entry application and immediately exit the menu. The ".." in the path to Popstan2020.pff tells
CSEntry to go one directory up, i.e. to go to the parent of the Menu directory and from there to Household/Popstan2020. pff.

Making the Entry Program Return to the Menu Program

When we exit the household application, we want to return to the menu but the menu program stopped when we launched
the household application. To get it to restart we add a parameter to the pff file to tell it to start the menu program when it
exits. Right click on the Popstan2020.pff and choose "Edit". This will bring up the pff file editor. From the Add menu add a
new On Exit PFF entry. Use the browse button in that entry to locate the Menu.pff which will fill in the path to it.

Deploying multiple applications on Mobile

When you copy your application to your phone or tablet it is important that you preserve the same folder structure as you
have on the PC. In our case we must create a separate menu folder to contain the pen and pff files for the menu and this
menu folder must be in the same parent directory as the Household folder so that when we use "../Household/" from the
menu application it points to the folder containing the entry application we are launching. When you have multiple
applications like we do, it can be time consuming to create the .pen files for each application and then copy only the .pen
and . pff files to the mobile device.

To simplify deployment use the Deploy Application tool from the Tools menu. This tool automates the process of publishing
and deployment. Drag the whole Popstan2020 folder to the Files window. The tool will add the entry applications (.ent files)
and . pff files found in the folder and its subfolders to the list of files to deploy. By default it does not add data files since data
files are often just used for testing. However, for our application we need to add the lookup files for the districts and
possession limits. We can do this by dragging the individual files onto the files window.

Once the files have been added, enter the name and an optional description for the application. Under Deploy To choose
Local Folder and click on "..." to choose the folder to deploy to. Click the Deploy button. This will automatically create .pen
files for each data entry application and copy them, along with any other files from the Files window, into the deployment
folder chosen. The deployment folder can then be copied directly to the mobile device. The folder structure of the
deployment folder will match the folder structure we are using for development and testing on the desktop so there will be
no problems with paths.

Popstan2020Census - C5Pro Deploy Application - O X

File Help

Package name |P0pstan 2020 Census |

Description |Examp|e census listing and household questionnaires fnJll
Files
=] Add fies...
=l | Household
. " Popstan2020.ent
= Data Add folder...

B3 PossessionLimits.csdb
-3 Districts.cadb
Listing
i M Listing ent
= | Menu
..M Popstan2020Menu ent

------ § Popstan2020Menu pff

Dictionaries to Upload For Synchronization

Popstan2020.dcf A
Districts.dof

PossessionLimits.dcf

Listing dcf v

Deploy To

(O CSWeb

() Dropbox

O FTP

(® Localfolder  |C:\Users\census'\Desktop'Popstan2020"DeployPopstan2020

) Zipfile

Deploy

Once you have set up the deployment the first time you can save the deployment specification file to simplify the process for
future deployments.

In the next session we will learn how to use this tool to deploy applications to a server for deployment over the Internet.

Tidying up the Menu Program

Currently the menu program shows up in the applications list on Android as "Menu" instead of something like "Popstan 2020
Census". In addition, when we tap on Menu we have to then tap "Start New Case" which doesn't make sense for a menu.
We can fix both of these problems by modifying the pff file for the menu. Right click on the Menu.pff and choose "Edit with
pff editor". Change "Start mode" to "Add" so that we won't have to tap "Start New Case". Enter "Popstan 2020 Census" for

Page 63 of 87 Session 08: Menu Programs



the description to change what shows in the application listing. While we are at it, open the Menu program in the CSPro
designer and in data entry options turn off the display of the case tree since the case tree is not useful for menu programs.
Finally, in the data entry options, check "Automatically advance on selection". This way the interviewer will not have to tap
next to move from one field to another in the menu program.

Now that we have the menu program to launch the household data entry program, we don't want the household
application to appear in the list of applications on the mobile device. By default, CSPro creates the list of applications from all
of the pff files in the CSEntry directory on the device. To prevent an application from being listed you can edit the pff file in
the pff editor and choose "Hidden by Default" or "Never" for "Show in Application Listing".

Creating a Dynamic Menu

When the interviewer launches the data entry application, they can currently enter any ID items but we would like to
restrict them to only interviewing households that have already been listed. We can do this by displaying the households
from the listing file in a dynamic value set and having the interviewer choose which one to interview.

In order to read the listing file, we need to add the listing dictionary as an external dictionary in the menu application. We
will also need to add a new menu item to list the households. This will be a new variable in the dictionary named
CHOOSE_HOUSEHOLD. We will make it 10 digits long to hold the full set of ID items (province, district, EA, area type and
household number). It will be an alpha field as this will be easier to work with later on. The interviewer menu will need to be
modified to skip to this new field instead of calling launchHouseholdDataEntry().

PROC INTERVIEWER_MAIN_MENU
postproc

// Handle the menu choice
if $ = 1 then
// List households
launchHouseholdListing();
elseif $ = 2 then
// Household questionnaire
skip to CHOOSE_HOUSEHOLD;
elseif $ = 9 then
// Logout
stop(1);
endif;

// Show interviewer menu again
reenter;

In the onfocus proc for CHOOSE_HOUSEHOLD we need to loop through every case in the listing file. We can use the
forcase statement to loop through the cases in the listing file and build the value set:

PROC CHOOSE_HOUSEHOLD
onfocus

ValueSet string householdVSet;

// Loop through all cases in listing file
// to build dynamic value set.
forcase LISTING DICT do
// Values are household ids concatenated together
string code = maketext("%v%viviv",
LTI _PROVINCE, LI _DISTRICT, LI_ENUMERATION_AREA,
LI_HOUSEHOLD_NUMBER);

// Labels have househould number and name of head
string label = maketext("%v: (%s)", LI_HOUSEHOLD_NUMBER,
strip(LI_NAME_OF_HEAD_ OF_HOUSEHOLD));

householdVSet.add(label, code);
endfor;

setvalueset($, householdVSet);
When we run this we should see a list of the households in the listing file as the value set for CHOOSE_HOUSEHOLD.

While the dynamic value set works correctly when there are households that have been listed, if the listing file is empty we
get an empty value set. Instead we should show an error message and return to the main menu. We can determine if the
value set is empty by checking the length of the list of codes in the value set. You can access the lists of codes and labels of a
value set using the variable name followed by a dot (".") and "codes" or "labels". The codes and labels properties of the
value set are CSPro lists so you can retreive their sizes using the length() function.

PROC CHOOSE_HOUSEHOLD
onfocus

ValueSet string householdVSet;
string code, label;

// Loop through all cases in listing file
// to build dynamic value set.
forcase LISTING DICT do
// Values are household ids concatenated together
code = maketext("v%viviviv'",
LTI _PROVINCE, LI _DISTRICT, LI_ENUMERATION_AREA,
LI_AREA TYPE, LI_HOUSEHOLD NUMBER);

// Labels have househould number and name of head
label = maketext("%v: (%v)", LI _HOUSEHOLD NUMBER,
strip(LI_NAME_OF HEAD OF HOUSEHOLD));

householdVSet.add(label, code);
endfor;

if length(householdVSet.codes) = 0 then

errmsg("No households have been listed. Please list households first before
interviewing.");

reenter INTERVIEWER_MAIN_ MENU;
endif;

setvalueset($, householdVvSet);

Setting the "Key" attribute in the pff

The next step is to handle the menu choice in the postproc to launch the household data entry program using the ID items
for the household that was chosen. In order to do this, we will need to pass the ID items of the selected household to
household application as the "Key" attribute of the pff file. The household data entry program will then automatically read
the value from the pff and prefill the ID items once it is launched. To set the "Key" attribute in the pff file call the
setproperty() function of the pff variable and pass it the attribute name and attribute value. The value for the key should be
the entire caseid string picked in the field CHOOSE_HOUSEHOLD.

// Launch household questionnaire data entry application

function launchHouseholdDataEntry()
Pff householdPff;
householdPff.load("../Household/Popstan2020.pff");
householdPff.setProperty("Key", CHOOSE_HOUSEHOLD);
householdPff.exec();

end;

CSEntry prefills the ID items using the string passed in the "Key" attribute. Let's also make the field protected if it is prefilled.
We can do this using protect.

PROC PROVINCE
preproc

// protect the field if it is prefilled
if visualvalue(PROVINCE) <> notappl then

// protect field so the interviewer cannot modify it
protect (PROVINCE, true);

endif;

The logic for DISTRICT, ENUMERATION_AREA, and HOUSEHOLD_NUMBER is similar.

Using a Lookup File for Login

Rather than have the user choose their role, lets assign each interviewer and supervisor a staff code and then create a
lookup file containing the staff codes along with the role and the district/enumeration area assigned to the
interviewer/supervisor. Here is an example staff file:

Staff code Name Role (1=interviewer, 2=supervisor) Province District EA
001 Shemika Rothenberger 2 1 1

002 Andrew Benninger 1 1 1 1
003 Angelica Swenson 1 1 1 2
004 Zelma Hawke 1 1 1 3
005 Willis Catron 1 1 1 4

Note that for the supervisor we leave the enumeration area blank since they are assigned to an entire district and not to an
enumeration area. The supervisor will supervise all of the enumerators in their district.

Let's create a new external dictionary in the menu program for this file and use Excel to CSPro to convert the spreadsheet to
a data file named staff.dat.

Let's modify the login field so that the user enters the staff code instead of picking the role. We will need to increase the size
of the field and delete value set. In the postproc we now need to look up the staff code in the staff file, validate it, and then
skip to the appropriate menu based on the role.

PROC LOGIN

// Verify staff code using lookup file

if loadcase(STAFF_DICT, LOGIN) = © then
errmsg("Invalid staff code. Try again.");
reenter;

endif;

// Go to the appropriate menu for the role chosen
if STAFF_ROLE = 1 then
skip to INTERVIEWER_MAIN_MENU;
else
skip to SUPERVISOR_MAIN MENU;
endif;

When we deploy our application we need to remember to add the staff data file to our deployment specification, otherwise
we will not be able to login.

Preserving Login when Returning to Menu

Currently when you launch the listing program and return to the menu, you have to enter the user code again. Let's save the
login code so that you only have to enter it once. We can do this using the commands savesetting() and
loadsetting(). These commands store and retrieve values in persistent storage. Values saved in the settings are
available even after CSEntry is closed and restarted, and they are available in all CSPro applications on the same device. We
will save the login code in the login postproc after validating it:

PROC LOGIN

postproc

// Verify staff code using lookup file

if loadcase(STAFF_DICT, LOGIN) = © then
errmsg("Invalid staff code. Try again.");
reenter;

endif;

// Save login so we do not have to enter it again
savesetting("login", maketext("%v", LOGIN));

// Go to the appropriate menu for the role chosen
if STAFF_ROLE = 1 then
skip to INTERVIEWER_MENU;
else
skip to SUPERVISOR_MENU;
endif;

Settings are always stored as alphanumeric so we need to use maketext to convert the numeric LOGIN code to a string.

In the preproc we will try to retrieve the login code from the settings and if it is not empty we will use it instead of asking the
user to enter the code.

PROC LOGIN

preproc

// Check to see if there is an existing login code

// use that

if loadsetting("login") <> "" then
LOGIN = tonumber(loadsetting("login"));
noinput;

endif;

Finally, we need to clear the setting on logout:

PROC INTERVIEWER_MAIN_MENU
postproc

// Handle the menu choice
if $ = 1 then
// List households
launchHouseholdListing();
elseif $ = 2 then
// Household questionnaire
launchHouseholdDataEntry();
elseif $ = 9 then
// Logout
// Clear login from settings
savesetting("login", "");
stop(1);
endif;

// Show interviewer menu again
reenter;

PFF Parameters
The parameter section in the pff file allows you to pass any value to your data entry program. Let's use this to pass the
interviewer code chosen in the menu program to the household entry program.

// Launch household questionnaire data entry application

function launchHouseholdDataEntry()
Pff householdPff;
householdPff.load("../Household/Popstan2020.pff");
householdPff.setProperty("Key", CHOOSE_HOUSEHOLD);
householdPff.setProperty("INTERVIEWER CODE", maketext("%v", STAFF_CODE));
householdPff.exec();

end;

The next step is to modify the household data entry application to prefill the interviewer code using the parameter in the pff
file. This can be done using the sysparm() command which retrieves a parameter by name from the pff file. We can do
this in the preproc of the INTERVIEWER_CODE in the household application. sysparm always returns the result as a string
so we need to convert it to a number.

PROC INTERVIEWER_CODE
// Prefill with parameter from pff
preproc
if sysparm("INTERVIEWER CODE") <> "" then
$ = tonumber(sysparm("INTERVIEWER CODE"));

// make field protected
protect($, true);
endif;

The Completion Report
Let's add the report to the supervisor menu. We will create a report that shows the total number of households by interview
status. Here is an example:

Completion Report

Province: 01 District: 01

Interview Status
Completed: 10
Non-contact: 1
Vacant: 2

Refused: 1

Partially complete: 5
Total: 19

To generate this report, we need to loop through all the cases in the household dictionary and count the number of cases
with each interview status. We can use forcase to do this. To compute the totals for each category we create a local
variable for each status and increment the appropriate variable every time we encounter a household with that status.

// Display the completion report that shows total interview status

// for all cases in the supervisor's district.
function showCompletionReport()
string reportFilename = maketext("%sreport.txt", pathname(Application));

File tempFile;

setfile(tempFile, reportFilename, create);

filewrite(tempFile, "Completion Report");

filewrite(tempFile, "----------------- ");

filewrite(tempFile, "");

filewrite(tempFile, "Province %v District %v",
STAFF_PROVINCE,
STAFF_DISTRICT);

filewrite(tempFile, "");

filewrite(tempFile, "Interview Status");

filewrite(tempFile, "---------------- ");

numeric complete = 0;

numeric nonContact 0;

numeric vacant = 0;

numeric refused = 0;

numeric partial = 9;

forcase POPSTAN2020 DICT do

if INTERVIEW_STATUS = 1 then
complete = complete + 1;

elseif INTERVIEW_STATUS = 2 then
nonContact = nonContact + 1;

elseif INTERVIEW_STATUS = 3 then
vacant = vacant + 1;

elseif INTERVIEW_STATUS = 4 then
refused = refused + 1;

else
partial = partial + 1;

endif;
endfor;
filewrite(tempFile, "Completed: %d", complete);
filewrite(tempFile, "Non-contact: %d", nonContact);
filewrite(tempFile, "Vacant: %d", vacant);
filewrite(tempFile, "Refused: %d", refused);
filewrite(tempFile, "Partially complete: %d", partial);
filewrite(tempFile, "Total: %d", complete + nonContact +

vacant + refused + partial);

close(tempFile);

view(reportFilename);

end;

Exercises

1. Implement the function launchHouseholdListing() in the menu program. It should write out and launch a pff file to run
the listing program. Set the "Key" attribute in the pff file to the province, district, and enumeration area. Set the
staff area type and the code from the staff file as a parameter in the pff file. Prefill the staff code in the listing
program. Make sure that the listing program returns to the menu program on exit.

2. Modify the logic in the onfocus and postproc of CHOOSE_HOUSEHOLD to add an extra option to the end of the list
labeled "Back" that will move back to the interviewer main menu.

3. Modify the dynamic value set we create in the onfocus proc of CHOOSE_HOUSEHOLD to show the interview status
(A10 in the household questionnaire) in addition to the household number and the head's name. To do this you will
need to use loadcase on the household questionnaire dictionary to find the case from the household data file. You will
also need to handle the case where the household does not yet exist in the household data file.

4. Create a new menu called "Summary Reports" that has options for the completion report (the one we implemented
already) and a new "Total Population Report" that you will implement. This new menu should also have an option to
go back to the main menu. The "Summary Reports" menu should be accessed from the Supervisor Main Menu. The
new "Total Population Report" should look like:

Total Population Report

Province: 01 District: 02

Male: 1020
Female: 1025
Total: 2045

Session 09: Synchronization

At the end of this session participants will be able to:

e Deploy applications to mobile devices over the Internet
e Use the synchronization options dialog to add synchronization to an application
e (Create advanced synchronizations using CSPro logic to synchronize data files and update applications in the field

e Create peer to peer synchronizations between devices for situations where there is no Internet access

Synchronization in CSPro

After collecting data in the field, you need to get the data from the interviewer's devices back to headquarters to create a

combined data file. While this can be done manually by connecting each device to a laptop, copying the individual data files
and combining them using the Concatenate Data tool, this is rather inconvenient for large surveys. Instead, it is possible to
send the data from each device over the Internet to a central server that will combine the data into a single data file to be

used for further processing. In CSPro this is called synchronization.

Server

Interviewer Interviewer Interviewer

Direct synchronization between interviewers and central server over the internet

The Sync Server

In order to use Internet synchronization in CSPro you need a server accessible over a network. CSPro supports three types of
Synchronization servers:

o CSWeb: A free web application that can runin a local data center or hosted on a server in the cloud. Best for large
surveys. Requires skills in setting up and administering a website and SQL database.

e Dropbox: A free service for synchronization with servers in the cloud using an account created at www.dropbox.com.
Data is stored on the servers managed by Dropbox in the United States. Best for small surveys when the skills and
infrastructure for setting up CSWeb are not available.

e FTP: CSPro can synchronize with any FTP (File Transfer Protocol) server accessible over the network. Best for small
surveys when the skills and infrastructure for setting up CSWeb are not available and you don't want your data on the
Dropbox servers.

For our examples in this workshop we will use a CSWeb server, however using Dropbox or FTP is very similar. Installing the
CSWeb server is beyond the scope of this training so we will work with a server that is already installed and running. The

process for setting up a CSWeb server is well documented in the CSWeb User's Guide.

Deploying Applications

Instead of copying our application to a mobile device using a USB connection we can download it from a server. Before we
do that we need to use the Deploy Application tool to upload the application to the server. We saw in session 8 how to use
the Deploy Application tool to deploy to a local folder. To deploy to a server simply change the "Deploy To" to the type of
server to upload to. You can deploy to a CSWeb server, Dropbox or FTP. The deploy tool will upload all of the files that you
have dropped onto the files list to the server. It will also upload some or all of the dictionaries in your application to the
server. Uploading a dictionary to the server is required before you can synchronize the data file associated with that

dictionary.

Once you have uploaded the application to the server you can download to a mobile device by choosing "Add Application"
from the menu in the applications listing screen. This will ask you for the server details and then display a list of applications
available on the server. Once you download an application it is copied to the CSEntry directory of your device.

You can also deploy updated versions of your application to the server and use the same procedure to download the updated

application to your device.

Adding Synchronization to the Data Entry Application

In order to use synchronization, you must first add the server information to the application to enable synchronization. In
CSPro choose "Synchronization" from the Options menu. Choose the server type (CSWeb, Dropbox, or FTP). For CSWeb or
FTP, enter the server URL (for Dropbox no URL is required). You can use the "Test Connection" button to verify that the URL
is correct. For CSWeb, the URL to use for synchronization in CSPro will usually end with "/api" while the URL to use to access
the web interface will usually end in "/ui". The former is for CSPro itself to connect to the server while the latter is for use

only in the web browser.

Synchronization Options

Enable simple synchronization
Synchronization Service
Service: 0 csweb

Dropbox
FTP
Local Files

Custom

LRL: https: /fwww . my-csweb-server.org/api

Synchronization Direction

IUpload changes to server (PUT) e

OK

In the dropdown, you have three choices for how to synchronize the main data file:

Test Connect

Help

e Upload changes to server: all new cases and any cases modified locally will be uploaded to the server.

e Download changes from server: all new cases and cases modified on server will be downloaded from server.

e Sync local and remote changes: combines the first two options by sending updates to the server and downloading

updates from the server.

In most cases you will want to choose either the first or third option. Use the third option if you want to share cases between
multiple devices in the field as this will download every case on the server onto the local device. For a survey operation with
a large number of interviewers you may want to avoid this as it will cause you to download a lot of data. For our test, we will
use the first option.

Once the sync settings are complete, regenerate the pen file. The sync settings are stored in the pen file so if it is not
updated you won't be able to sync. Copy the pen and pff file into the Popstan2020Census directory under the files directory
on the server so that it will be downloaded to the local device during sync.

Running Sync
To synchronize the application with the server, run the application in CSPro and from the case listing screen choose

"Synchronize..." from the file menu on the PC or tap the synchronize icon () on mobile.

Downloading the Combined Data File

Once you have used synchronize to upload data to the server you can download the data file from the server using the Data
Manager tool. This will download a single data file that combines all the data uploaded from all the devices that have
synchronized with the server. Open Data Manager, choose the server type, enter the same URL you used in the
synchronization options dialog, click "Show Available Data" to download the list of data files available on the server, choose
the data file, choose a place to save the downloaded file and click "Download".

Download Data From Synchronization Service

¥ou can create a new data source by downloading data from a synchronization service, Provide the details of the service and dick Show Available
get a list of downloadable data.

Synchronization Service
Service: 0 csweb

Drophbox
FTP
Local Files

Custom

URL: https: /v my-csweb-server . org/api

Show Available

Data on Synchronization Service

Label MName Cases (Mot Deleted)
Popstan2020 POPSTAM2020_DICT 5

Local Data Source

C:\MyData.cedb

Download Cancel Help

Using the CSPro Web Server interface

The CSWeb server has a web interface that you can access through a browser. The URL for the web interface is similar to the
one used in CSPro except that the "api" at the end is replaced by "ui". The web interface allows you to manage deployed
applications, data dictionaries and user accounts.

C5Web

c @ @ cs ui/dashboard N D -

admin ® Logout

L Apps
File Cases Download Delete
Popstan2020 0 £ 2 Tl
Listing 0 & 1)
Simple CAPI Dictionary 102 & Tl

If you are using CSWeb, you can avoid filling out the download dialog by clicking on the "Download" icon in the CSWeb
interface in the browser. This will download a pff file. When you open the downloaded pff file, it will automatically open Data
Manager and begin downloading the combined data. Once you have downloaded the data file the first time, you can use the
"synchronize" function in Data Manager to only download the newly updated data rather downloading the entire file a
second time. If you are using Dropbox or FTP, CSPro will automatically create a pff file named DownloadData-<your
dictionary name>. pff in your Dropbox or on your FTP server that will launch Data Manager to download the data.

Group exercise — Sync Race
Split into groups of 3-4 people and create a small CSPro application for the following questionnaire:

Synchronization Exercise Questionnaire

A) Team name:

B) Names of team members:

|| B | =

Each team should use their team name as the name for their application's dictionary. If two teams use the same
dictionary name they will both sync to the same data file on the server and we would like to avoid that.
Deploy your application to the following server:

URL: http://csweb.teleyah.com/api

Admin username: test

Admin password: password
Add synchronization to your application using the above server details.

Choose to upload data to the server.

Do NOT sync the application files (pen and pff).
Run your application, complete a case and synchronize to send your data to the server.
The first team to get their data uploaded to the server is the winner.

Synchronization from Logic

Sometimes we need more control over the synchronization than the sync options dialog provides. For example, if there are
external dictionaries in an application, if you want to sync with a universe or if there are multiple applications to sync at
once. CSPro provides logic functions to run synchronizations from within an application that allow more complex
synchronizations.

Let's add synchronization to our menu application that will upload the main data file and download the latest version of the
pen and pff files for the menu application and for the listing and household questionnaires. We will add a new entry in the
supervisor main menu called "Synchronize with Headquarters".

The first step in running sync from logic is to call the command syncconnect () and pass it the server information. If this
function succeeds then we are connected to the server and we can make calls to the commands syncdata() and
syncfile() to synchronize data files and non-data files respectively. Finally we call syncdisconnect () to end the
session.

The commands syncdata() and syncfile() operate differently. syncdata() may only be used with data files in
CSPro DB format. With syncdata( ), CSPro keeps track of which cases in the file have already been uploaded to the server
so that it avoids transferring cases that have already been synced. This reduces the amount of data transferred and keeps
data costs and transfer times to a minimum. It also avoids one interviewer overriding changes made by another interviewer
when they are working on different cases in the same file. syncfile () may be used with any type of file including text
files, images and application files (pen and pff). It does not look at the file contents so it simply uploads or downloads the
entire file, overwriting any existing version. When syncing data files, you should always use syncdata().

syncdata() takes two arguments: the direction and the dictionary name. The direction can be PUT (upload data to
server), GET (download data from server) or BOTH (upload and download). These options correspond to the drop down in
the sync options dialog. The second argument is the name of a dictionary in the application that corresponds to the data file
to synchronize. Note that this dictionary must be added to the application as an external dictionary.

In our application, we want to synchronize the household questionnaire dictionary so we need to first add it to the menu
program as an external dictionary. Now we can use it in the call to syncdata().

We will also add callsto syncfile() to download the latest versions of the application programs from the server. This
function takes the direction (PUT or GET), the "from" directory and the "to" directory. In the case of GET the "from"
directory is the path on the server and the "to" directory is the local directory on the device.

// Upload cases to web server at headquarters
// and download latest version of application files
function syncWithHeadquarters()

// Connect to the webserver
if syncconnect(CSWeb, "http://csweb.teleyah.com/api") then

// Sync cases from household data file with server
syncdata(BOTH, POPSTAN2020_DICT);

// Download latest version of menu application
syncfile(GET, "/Popstan2020/Menu/Menu.pen”, "./Menu.pen");
syncfile(GET, "/Popstan2020/Menu/Menu.pff", "./Menu.pen");

// Get latest household data entry program
syncfile(GET, "/Popstan2020/Household/Popstan2020.pen”,
"../Household/Popstan2020.pen");

// Disconnect from the server
syncdisconnect();

endif;
end;

Now that we have added the synchronization logic to the menu we rebuild the pen files and copy them to the files directory
on the server. This will allow us to run the menu program and test our synchronization.

Group exercise — Sync Race Il
In your teams modify the sync exercise questionnaire to take a photo of the team and save it to a file.

Synchronization Exercise Questionnaire

A) Team name:

B) Names of team members:

1

2

3

4
C) Team Photo: 1 Yes 2 No
D) Synchronize with server: 1 Yes 2 No

Add two questions, C and D, to your application. For question C, take a photo if the interviewer answers "yes". For
question D, add logic to the postproc of the question so that it synchronizes with the server if the user selects "yes".
Use syncfile () with PUT to upload the photo to the directory /photos/ on the server.

Run your application, enter a case and synchronize to send your photo to the server.

The first team to get their photo uploaded to the server is the winner.

Security Considerations

It isimportant to ensure the security of your server and data. Computer and network security is a challenging problem and
well beyond the scope of this training. Here a few security considerations to think about. For any large survey or census
operation you should consult with a security expert to ensure that your data is safe.

Use HTTPS on the web server for data transfer

This encrypts the data transmission that goes over the Internet. It requires purchasing and installing an SSL certificate on the
server. Without SSL, server passwords are sent unencrypted and are vulnerable to hackers.

Use a private network

For censuses and large surveys you may be able to work with the telecom provider to create a private network that only
your devices can access. This avoids putting your server on the public Internet where it could be vulnerable to attacks. This
also makes it possible to limit your devices to only be able to access your server and not other websites that you don't want
your interviewers to use.

Use device encryption on Android devices

By setting a PIN code on your Android tablet or phone you enable encryption of files on the device using strong hardware
encryption. This makes it very difficult for anyone without the PIN code to retrieve data on the device.

Consider how to manage passwords

It is more convenient to have a single password that is shared by all devices and is hardcoded in the pen file however it is
more secure to use a different username/password for each interviewer and have them enter it for every sync. For a large
number of interviewers managing passwords and resetting passwords for those who have forgotten their passwords could
be a significant management burden. You will need to find the right balance of security and convenience. If you want to
hardcode the password in your pen file you can pass the username and password in the call to syncconnect (). If you do
not provide a username and password in this call, the interviewer is prompted to enter the password the first time they sync
with the device. After that the username and password are saved on the device so they do not need to be entered for
synchronizations. If you don't want the password saved on the device, you can use the prompt () function to have the

interviewer enter the password before each sync and then pass the password received from the call to prompt () to
syncconnect().

Synchronization with a Universe

The syncdata() command has an optional third parameter, the universe. This is a string that CSPro tries to match to the
concatenated case ID items to limit the data transferred. If a universe is provided, then only cases who case IDs match the
universe are synchronized. For example, to limit synchronization to only province 1, district 2 we would call:

syncdata(BOTH, HOUSEHOLDQUESTIONNAIRE_DICT, "102");

This will only synchronize cases whose case IDs start with "102". This limits the sync to cases in province 1 and district 2.
Similarly, if we used "1" as the universe all cases in province 1 would be synced.

This is useful when there are a large number of interviewers and you don't want every interviewer to download the data
from every other interviewer. You can specify the geographic area assigned to the interviewer as the universe so that only
cases in the interviewer's area are synced.

Peer to Peer Synchronization Using Bluetooth

In some places, interviewers do not have reliable Internet access and cannot synchronize directly with the server. In this
case, you can use peer to peer to synchronization. Interviewers synchronize with a supervisor's tablet or laptop over
Bluetooth. Later, the supervisor goes to a location where they can connect to the Internet and they synchronize with the
central server.

Server

Bluetooth N < Bluetooth
Supervisor
Bluetooth
Y v
Interviewer Interviewer Interviewer

Synchronization with supervisor over Bluetooth when interviewers do not have internet

In order to implement Bluetooth synchronization, we need to add logic on both the supervisor device and on the interviewer
device. One device acts like the web server, processing synchronization requests from the client device, and the other
device, the client, sends syncfile and syncdata commands the same way that it would to a web server. For our
example let's make the supervisor device be the server and the interviewer device be the client. This is arbitrary, we could
easily reverse the roles.

To extend the menu program to support peer to peer synchronization we will add a new menu option to the supervisor main
menu "sync with interviewer" and a new menu option to the interviewer main menu "sync with supervisor".

The logic on the interviewer's tablet is nearly identical to the logic used for web synchronization. The only differences are the
arguments to syncconnect (). When using Bluetooth, syncconnect () does not need a URL, username or password.

syncconnect(bluetooth)

syncconnect () will scan all nearby Bluetooth devices and present a list to the interviewer who can choose which device
to connect to. Alternatively, if you know ahead of time the name of the device to connect to you may specify it as a second
parameter. In this case syncconnect () will try to connect directly to the named device.

syncconnect(Bluetooth, "Supervisore3")

The rest of the logic is the same as for web synchronization except that the "from" paths change to match the paths on the
supervisor device.

function syncWithSupervisor()
// Connect to the supervisor device
// We do not specify the device name to connect to which allows the interviewer
// to pick the device from a list of nearby devices.
if syncconnect(bluetooth) then

// Sync main data file.
syncdata(BOTH, POPSTAN2020 DICT);

// Download latest application files from the supervisor. The
// root file on the supervisor tablet is in the Menu folder.
syncfile(GET, "./Menu.pen", "./Menu.pen");

syncfile(GET, "./Menu.pff", "./Menu.pff");

// Since the current application is in the Menu folder we need to use

// "../Household" to go up one level and back down into the Household

// folder for the household application files.

syncfile(GET, "../Household/Popstan2020.pen",
"../Household/Popstan2020.pen");

syncdisconnect();
endif;
end;

The logic for the supervisor is even simpler. We simply call the command syncserver () which runs the Bluetooth server
and waits for connections.

function syncWithInterviewer()
// Run the Bluetooth server to receive data from interviewer.
syncserver(Bluetooth);

end;

By default, the root directory is the application directory on the server.

Group exercise — Sync Race Il
In your teams modify the sync exercise application to support Bluetooth sync.

Synchronization Exercise Questionnaire

A) Team name:

B) Names of team members:

AIWIN| =

C) Team Photo: 1Yes 2 No

D) Synchronize: 1 Web server 2 Supervisor 3 Interviewer 4 None

Modify question D so that you have the option of synchronizing via Bluetooth.

Run your application on two devices, complete a case on one device and synchronize to send your photo to the other
device via Bluetooth. Use the second device to synchronize to send your photo the web server.

The first team to get their photo uploaded to the server is the winner.

Exercises

1. Modify the syncWithHeadquarters() and syncWithSupervisor() functions in the menu program to also sync the listing
data file and the staff lookup file. The listing file should be synced using BOTH and the staff file should be downloaded
from the server using syncdata with GET.

2. Modify the syncWithHeadquarters() and syncWithSupervisor() functions so that only data in the geographic area
assigned to the interviewer/supervisor are synced. For the interviewer this means that only households in the
assigned EA are synced and for the supervisor only households in the assigned district are synced.

Session 10: Batch Edit and Export

At the end of this session participants will be able to:

e Implement consistency checks in a batch edit application
e Use batch edit to add calculated variables and recoded variables
e Use batch edit to convert checkbox values to yes/no variables for analysis

e Use the Export Data tool to take data from CSPro to other software packages

Creating a Batch Edit Application

A batch edit application is like a data entry application but without the forms. It is meant to be run after data entry to
detect and fix problems in the data file. A batch edit application takes an input data file and runs logic onit. It generates
both a report, called a listing file, and optionally an output data file which is a modified version of the input file. A batch edit
application never changes the input file.

To create a batch edit application you choose File->New from CSPro and choose Batch Edit Application. You then choose a
dictionary. This is usually the same dictionary that you used for data entry.

The user interface for working with batch applications is similar to the one for working with data entry applications except
that there are no forms. Instead there is a tab in the tree one left side of the window for edits. Just like in data entry, you
add logic to PROCS. Instead of running interactively, all the error messages are written out to a log file for review after the
whole program has run.

Checking for Errors
To add consistency checks we proceed just as we did in our data entry application by adding logic to the appropriate PROC.
Let's start with a simple check that the age of first marriage is not greater than the age.

PROC AGE_AT_FIRST_MARRIAGE

// Check for age at first marriage less than current age
if AGE_AT_FIRST_MARRIAGE > AGE then

errmsg("Age at first marriage greater than age");
endif;

Next we run the application but first we need a test data file. You can use the file Popstan2020Raw.csdb. Run the
application against this test data. After the application runs, we see the log file. The log file reports that we have a case
where this error exists.

Process Messages

*** Case [3691141112] has 1 messages (0O E / O W / 10)
U -20 Age at first marriage greater than age

User unnumbered messages:

Line Freg Pct. Message text Denom

20 1 - Age at first marriage greater than age -

CSPRO Executor Normal End

To figure out what the problem is we can open up the problem case in data entry. The printout in the listing file contains the
case identifiers which we can use to find the case. You can copy the case ID from the listing file and use it with Find Case on
the Edit menu in CSEntry.

Getting Summary Information

If you add summary after the errmsg statement, the individual error cases will not be written to the listing file. With
summary, only the total number of times each error is encountered is written to the file. This can be helpful when working
with large data files that have a lot of errors making the listing file very large.

PROC AGE_AT_FIRST_MARRIAGE

// Check for age at first marriage less than current age
if AGE_AT _FIRST_MARRIAGE > AGE then

errmsg("Age at first marriage greater than age") summary;
endif;

Now when we run the application, the listing file no longer contains the individual cases where the errors occurred.

User unnumbered messages:

Line Freg Pct. Message text Denom

20 1 - Age at first marriage greater than age -
CSPRO Executor Normal End

If you add a denominator using the denom keyword, CSPro will calculate the percentage of cases where the error occurred.
The denominator to use depends on what is being counted. In our case it is the total number of household members with
age 10 or above and marital status married, divorced or widowed. We can calculate that total as part of our batch edit
program.

PROC GLOBAL
numeric numHHMembersl1@AndOverEverMarried = 0;

PROC AGE_AT_FIRST_MARRIAGE

if AGE >= 10 and MARITAL_STATUS in 2:4 then
numHHMembers10AndOverEverMarried = numHHMembers10AndOverEverMarried + 1;

// Check for age at first marriage less than current age
if AGE_AT _FIRST_MARRIAGE > AGE then
errmsg("Age at first marriage greater than age") summary
denom = numHHMembersl10AndOverEverMarried;
endif;
endif;

Now when we run the program we see that 0.2% of the individuals 10 and above who where ever married have an age at
first marriage greater than their age.

User unnumbered messages:

Line Freg Pct. Message text Denom

20 1 0.2 Age at first marriage greater than age 463

CSPRO Executor Normal End

Correcting Errors
In addition to using batch edit to find errors you can also use it to correct problems by modifying variables in your logic. Let's
simply cap the age at first marriage to never be greater than the age.

// Check for age at first marriage less than current age
if AGE_AT_FIRST_MARRIAGE > AGE then
errmsg("Age at first marriage greater than age. Capping age at first marriage at
age.");
AGE_AT_FIRST_MARRIAGE = AGE;
endif;

When we run this time we will specify an output file: Popstan2020Edited.csdb. The changes we make will only be made to
the output file. We can then rerun the batch application on the output file and make sure that we don't have any error
messages.

Instead of just assigning the value of AGE_FIRST_MARRIAGE we can use the impute command which does the assignment
just like "=" but also generates a nice report showing the values that were imputed.

// Check for age at first marriage less than current age
if AGE_AT_FIRST_MARRIAGE > AGE then
errmsg("Age at first marriage greater than age. Capping age at first marriage at
age.");
impute (AGE_AT_FIRST_MARRIAGE, AGE);
endif;

The imputation report will be opened in Text Viewer after you run the batch application but to see it you will need to go to
the Window menu and choose the file that endsin ".frg.Ist".

IMPUTE FREQUENCIES Page 1

Imputed Item AGE AT FISRT MARRIAGE: Age at first marriage - all occurrences

Categories Frequency CumFreq s Cum %
40 1 1 100.0 100.0
TOTAL 1 1 100.0 100.0

Adding Calculated Variables

It is often useful to add additional variables to your data file after data collection that are computed from the collected
variables. For example, let's add a yes/no/don't know variable to the household record that determines if the household is
headed by a child. First we add the new variable CHILD_HEADED_HOUSEHOLD to the dictionary (at the end of the housing
record so that we do not mess up our existing data). Then we add logic to the PROC of our new variable to impute the value.
A household is headed by a child if the age of the head is less than 18.

PROC CHILD_HEADED_HOUSEHOLD

// Set calculated variable child headed based on age of head
if AGE(1) < 18 then
// Head under 18, child headed
impute(CHILD_HEADED_HOUSEHOLD, 1);
elseif AGE(1) = 999 then
// Head age unknown
impute(CHILD HEADED_ HOUSEHOLD, 9);
else
// Head over 18, not child headed
impute (CHILD HEADED HOUSEHOLD, 2);
endif;

Run the program and look at the imputation report to see how many child headed households are in our data set.

Converting Checkboxes to Yes/No

The checkbox makes a nice interface but the resulting data is a string that is hard to interpret and deal with in other
software. To make it easier to use for export we can convert from the checkbox to a repeating item with yes/no options.
Let's convert the LANGUAGES_SPOKEN field to a series of yes/no variables. Create new yes/no items for each language at
the end of the individual record. Add value sets with Yes—1 and No—2.

For each of these new variables we want to set the value to yes if the corresponding language is checked in
LANGUAGES_SPOKEN. What function do we use to determine if an item in a checkbox field is checked? As before we use

pos().

PROC LANGUAGE_ENGLISH_SPOKEN
if pos("A", LANGUAGES_SPOKEN) > © then
LANGUAGE_ENGLISH_SPOKEN = 1;
else
LANGUAGE_ENGLISH_SPOKEN = 2;
endif;

PROC LANGUAGE_FRENCH_SPOKEN

if pos("B", LANGUAGES_ SPOKEN) > © then
LANGUAGE_FRENCH_SPOKEN = 1;

else
LANGUAGE_FRENCH_SPOKEN = 2;

endif;

PROC LANGUAGE_SPANISH_SPOKEN
if pos("C", LANGUAGES_SPOKEN) > © then
LANGUAGE_SPANISH_SPOKEN = 1;
else
LANGUAGE_SPANISH SPOKEN = 2;
endif;

The remaining languages follow the same pattern.

The Export Data Tool

The CSPro Export Data tool is available from the Tools menu. When you first start Export Data you are prompted to provide
a data dictionary. Choose the Popstan2020 dictionary. From the main export screen, you can choose which records/variables
to export using the checkboxes next to each one. To start with let's just pick the ID items and the first few fields from section
F. At the bottom of the export window you can choose the file format to export to. For this exercise we will choose CSV
which can be easily opened in Excel. To run the export, click on the traffic light. You are prompted to choose the data file.
We will use the Popstan2020Edited.csdb data file. Finally, you are prompted for the name(s) of the exported files. Once the
export completes the exported files are displayed in Text Viewer. Let's open them in Excel and see what we have. Note that
each household is saved in a separate line in the Excel file.

ENUMERATI HOUSEHOLD
PROVINCE DISTRICT ON_AREA AREA TYPE _NUMBER FO1 F02 FO3
2 14 214 1 1 3 1 3
3 27 301 1 1 9 9 1
1 1 345 1 5 1 1 1
3 26 222 2 2 1 4 1
4 37 422 2 1 1 5 1
2 2 214 3 1 1 2 4
1 1 101 1 1 2 5 1

Exporting Repeating Records

Let's try exporting a few fields from section B: line number, name and sex. Notice for each variable we exported we get 50
columns. Why is that? We are getting one column for each record occurrence. CSPro is still exporting each household on a
single row and since there are up to 50 people in the household it generates 50 columns for each variable. Even empty
occurrences are still generating columns in the spreadsheet.

lrow=1

Bl |BL |BL |BL |BL |BL |BL |BL |BL |BL |BL |BL |BL |BL B household
01 |02 |03 |04 |05 |06 |07 |08 |09 |10 |11 |12 |13 |14 |15

1 v

1| 2| 3 <~

2

1] 2] 3] a

1| 2] 3] a

1] 2] 3

1] 2] 3| a| s| 6| 7

Having a column for each occurrence can complicate working with the data. For example, in this file it is rather tough to do
something as simple as count the total number of people. If instead of choosing the default setting of putting multiple record
occurrences in a single row (the All in One Record setting) try selecting As Separate Records. Now we get each person in the
household in a separate row. It is important to include the household ID items when doing this so that it is clear which
people are in which household.

lrow=1

/—person

PROVIN | DISTRIC | ENUMERATION_A | AREA_TY | HOUSEHOLD y(n B

CE T REA PE BER 1 |B2 B3 | BS
2 14 | 101 1le—"" 1| 1| John 1|27
2 14 | 101 1 1| 2| Mary 2| 26
2 27 | 200 1 1| 1|Jane 2| 40
2 27 | 200 1 1| 2| Billy 1] 13
2 27 | 200 1 1| 3| Tina 2| 13

Exporting Multiple Record Types

Now let's try exporting the first few items from both the person record (B) and the deaths record (E). With our current
settings, Export Data warns us that only items from the first record will be exported. Why? The problem is that if you put
each record in its own row then some rows would have household members on them and others would have deaths on them
but then the columns would not match.

The solution is to export each record in a separate file by selecting Multiple Files (one for each record type) under Number of
Files Created. Doing this generates two files: PERSON.csv which contains the household members and DEATHS.csv which
contains the deaths.

PERSON.csv
PROVINC ENUMER HOUSEH
E ATION_A | AREA_TY | OLD_NU
el e PE MRER _ | B1 B2 B3 BS
—> | 2 14| 101 1 1 1| John 1 27
2 T2 I0T T 1 2 | Mary 2 26
2 27| 200 1 1 1| Jane 2 40
2 27 200 1 1 2| Billy 1 13
2 27| 200 1 1 3| Tina 2 13
DEATHS.csv
ENUMERA HOUSEHO
TION_ARE | AREA TYP | LD_NUMB
PROVINCE |_DISTRIGT—|-A E ER E3 E4 ES
—> < 2 14 | 101 1 1 1| Erwin 20140211
2 T2 T0T T 1 2 | carmine | 20161201
2 27| 200 1 1 1 [ Arnold 20150113

The households in these two files can be linked together based on the ID items.

It also possible to have Export Data join together single and multiple records. If you select this option, then one file will be
generated for each record with multiple occurrences and the selected columns for all the selected singly occurring records
will be added to each row in each of the exported files. For example, if we choose the tenure status (FO6) and type of main
dwelling (FO3) from the single record housing and also choose the first few items from the person record then FO6 and FO3
will be added to each of the two exported files: PERSON.csv and DEATHS.csv.

PERSON.CSV
PROVINC | DISTRIC | ENUMERATION | AREA | HOUSEHOLD | FO3 FO6
E T _AREA _TYPE | NUMBER Bl B2
2 14 | 101 1 1 1 1 1| lohn
2 14 | 101 1 1 1 1 2 | Mary
2 27 | 200 1 1 2 3 1] Jane
2 27 | 200 1 1 2 3 2 | Billy
2 27 | 200 1 1 2 3 3 | Tina
k J ( J\ Y J
Y [
Id items Single record  Multiple record
A ) A
{ DEATHS.CSV u \
ENUMER
PROVINC ATION_A | AREA_TY | HOUSEHOLD N
E DISTRICT | REA PE UMBER FO3 FO6 E3 E4
14 | 101 1 Erwin
14 | 101 2 Carmine
27 | 200 3|1 Arnold

It is not possible to join multiple records to other multiple records from export.

Note that in some cases it may be easier to do the export of the single and multiple records without joining and then do the
join in the software that you have imported the data into.

Importing Data into SAS, SPSS, Stata and R
When exporting data to statistical packages, CSPro generates both a data file and script to run inside the statistical software
itself to run the import. For details on how to run this script for each package, see the helps for Export Data and look under

"Loading Exported Data Into...".

Saving your Export Specification
You can save your export settings as CSPro export specification file (.exf). You can later double click on this file or open it
from the Export Data Tool to retrieve all the selections that were made.

Exercises

1. Modify the batch edit application to add a check for someone with relationship of spouse but marital status that is
not married. Print a message for each case found. This should be done in the batch edit application NOT in the data
entry application.

2. Modify the batch edit application to add a check that the total number of rooms (FO1) is greater than the number of
bedrooms (F02)

3. Modify the batch edit application to add a check that the type of main dwelling (FO3) is consistent with the total
numbers of each dwelling type in FO5. For example, if the main dwelling in FO3 is detached house then the total
number of detached houses in FO5 must not be zero.

4. Modify the batch edit application to convert the disabilities (B11) from alpha (used for checkboxes) to a set of
numeric yes/no variables. Create the new variables in the dictionary and write logic to set the value of each of the
new variables based on the selections in B11.

5. Using the test data, export the household members along with the ID items to the package of your choice (Excel,
SPSS, Stata, SAS or R). Use that package to determine the total number of people by sex (total, males, females) and
also the number of people by sex for district 01.

6. Export the deaths record and the person record along with the ID items into the package of your choice. You should
export all the records at once, not one by one. How many households have more than one death? How many
households have children under 5 years of age?

